
1 Vast stramien 

Bij het maken van elk rapport – of we dat nu met Crystal Reports of bijvoorbeeld Access doen - gaan we 

te werk volgens een vast stramien: 

 

1. Welke informatie wordt er gevraagd? 

2. Welke database(s) moeten we gebruiken? 

3. Welke tabellen/queries daaruit hebben we nodig? 

4. Welke relaties liggen er tussen tabellen? 

5. Welke velden selecteren we? 

6. Welke gegevens moeten we afleiden uit welke velden? 

7. Welke restricties passen we toe op welke velden of afleidingen? 

8. Hoe groeperen we? 

9. Welke sorteervolgorde kiezen we? 

10. Welke cumulatieve bewerkingen (totalen, gemiddelden, aantallen etc.) voor welke groepen van 

gegevens? 

11. Hoe maken we het rapport op? 

 

Hoe passen we nu dit stramien toe in Crystal Reports? Er zijn twee manieren: 

 

▪ we gebruiken de Report Wizard 

▪ we doen het helemaal zelf 

 

En uiteraard is het ook mogelijk met de Report Wizard het voorwerk te doen om vervolgens met de hand 

eventuele aanvullingen aan te brengen. 

 

In deze cursus gaan we het helemaal zelf doen. Deze werkwijze geeft volgens ons het meeste inzicht in 

de stapsgewijze opbouw van een rapport. Er is nog een tweede argument. De Report Wizard is van 

versie 8.5 naar 9 (en 10) in functionaliteit achteruit gegaan. Was het in versie 8.5 nog mogelijk vanuit 

een eenmaal gemaakt rapport terug te komen in de Wizard; in 9 en 10 kan dat niet langer: je moet daar 

op eigen kracht verder. 

 

In hoofdstuk 2 maken we een simpel rapport waarbij we bovengenoemde stramien tot en met stap 5 

doorlopen.  

2 Een simpel rapport van begin tot eind 

2.1 Simpel rapport 

We gaan een simpel rapport maken. Helaas komen zelfs bij een simpel rapport een aantal ingewikkelde 

handelingen kijken. Die nemen we gelijk mee in de uitleg. 

 

▪ We starten Crystal Reports; het volgende scherm verschijnt 

 

 

 

▪ We kiezen de optie As a Blank Report 

▪ En klikken vervolgens op Ok  

Het volgende scherm  verschijnt: 

 

 

 

We staan nu voor de keuze hoe we contact willen maken met de database. Daarvoor staan ons in 

Crystal Reports diverse technieken ter beschikking. De keuze die we maken hangt ook af van de 

database waarmee we gaan werken.  

 

Voor de hand liggende keuzes zijn ODBC, Ole DB en als Database file. In deze cursus werken we met 

een Access Database. Die kunnen we in alle versies rechtstreeks koppelen via Database files. 

 

 

 

Let op:  

Er zijn zoals gezegd nog diverse andere koppelingsmogelijkheden. Verschillende databases zijn 

bijvoorbeeld ook rechtstreeks te koppelen, zoals SQL Server en Oracle.  

Andere mogelijkheden zijn onder andere OLE DB, ADO, DAO, Crystal SQL query files. Het voert te ver 

deze alternatieven hier uit te diepen. In de praktijk is bovendien ODBC de meest gangbare. 

Koppelen via ODBC is binnen bedrijven gangbaar. Daarom gaan we een dergelijke koppeling maken, 

maar dan met behulp van Windows. In een bedrijfssituatie zal deze mogelijkheid veelal niet beschikbaar 

zijn. Alleen de systeembeheerder kan dan een dergelijke koppeling maken. 

 

 

 

Via de optie Add Command is het nu ook mogelijk rechtstreeks SQL commando’s in te voeren: 

 

 

 

▪ En dubbelklikken vervolgens op Orders, Orderinformatie, Producten, Klanten, Categorieën en 

Werknemers  

▪ en vervolgens op OK 

▪ We maximaliseren dit scherm  

▪ En klikken op clear links  

▪ En leggen handmatig de relaties tussen de diverse tabellen 

 

 

Let op:  

We verwijderen de relaties en leggen ze vervolgens zelf omdat Crystal Reports velden op basis van 

gelijkluidende namen koppelt! Vanaf versie 9 gebeurt automatische koppeling alleen op gelijkluidende 

veldnamen waarvan er tevens één een index heeft. Zo worden klanten en leveranciers gekoppeld via 

postcode en bedrijf. 

 

Vanaf versie 11 is het mogelijk via dubbelklikken op de tabelnaam de tabelnaam te laten samenklappen.  

 

 

▪ Dan dubbelklikken we net onder een willekeurige relatiepijl 

 

Het volgende scherm verschijnt: 

In beide gevallen kunnen we bijvoorbeeld kiezen voor de Left Outer Join; we zullen dit in deze cursus 

verder niet behandelen. 

 

▪ We klikken twee keer  op OK  

 

We komen dan bij het hoofdscherm van Crystal Reports. De schermen van de verschillende versies 

verschillen op details van elkaar. We komen daar later op terug. 

Willen we weer terug naar de Database Expert, dan klikken we op het tonnetje.  

 

 

 

In het bovenstaande scherm zien we de Field Explorer. Die kunnen we aan- en uitzetten via de knop 

.  

 

In de vorige scherm is de Field Explorer gedocked in de linker bovenhoek. Daar kan het weggesleept 

worden. Door rechts te klikken op het losse venster en Dock te kiezen zetten we het weer terug. 

 

▪ Denk eraan het rapport gelijk even op te slaan 

▪ Kies File  

▪ We zetten Save data with report uit 

▪ En klikken in dit menu op Options 

▪ We vinken eventueel Autosave Reports After aan 

▪ Bekijk de andere opties, ook op de andere tabbladen 

 

We keren terug naar de Field Explorer 

 

▪ We drukken op de + voor Database Fields en vervolgens op de + voor klanten  

 

En krijgen het volgende scherm 

▪ Klik met de rechter muisknop op Klanten 

▪ Zet Sort Fields Alphabetically aan 

 

 

 

Let op:  

Rechts klikken op een veld in de Field Explorer levert de optie Browse Data. 

 

 

 

Die optie geeft een inkijk op de data in een kolom. Daar zit evenwel wel een beperking op. Alleen de 

eerste 500 verschillende gegevens worden getoond. Alle volgende waarden worden weggelaten. 

 

 

▪ Zet Show Field Type aan 

 

 

 

We krijgen dan extra informatie over het type veld en de lengte. 

 

We kunnen de velden ook standaard in alfabetische volgorde zetten. We doen dit via het menu File  

options    database: 

Vanaf versie 2008 kunnen we dat aanzetten door in de Field explorer op de rechter muisknop te klikken 

en de betreffende optie te kiezen. 

 

▪ We slepen een aantal velden – bijvoorbeeld klantnummer, bedrijf en contactpersoon - naar de 

detailsectie van het rapport 

▪ En slaan het rapport weer even op 

▪ We bekijken het via de knop Print preview  

▪ En sluiten het rapport 

2.2 Opgave: Outer join 

We gaan een nieuw rapport maken op basis van de klanten en de orders. 

▪ Maak een nieuw rapport met de tabellen klanten en orders. 

▫ Leg een relatie op basis van klantnummer. 

▫ Klik op OK. 

▫ Voeg aan het rapport de velden klantnummer uit klanten en order-id uit orders toe. 

▫ Start het printvoorbeeld en kijk naar het aantal records. 

▫ Ga terug naar de Database Expert. 

▫ Kies eventueel het tablad links. 

▫ Dubbelklik op de relatie.  

▫ Kies voor left outer join (mits klanten.klantnummer voorop staat). 

▫ Klik op OK en refresh. 

▫ Kijk naar het aantal records (als het goed zit moet dit nu twee hoger zijn. 

▫ Om de twee extra records te zien, klik op de Record Sort Expert. 

▫ Sorteer op order-id: bovenaan komen dan twee lege order-id’s. 
▫ Sla het rapport onder naam oefening0201.rpt. 

3 Bestaand rapport koppelen aan een andere database 

3.1 Updaten van de koppeling 

In de praktijk komt het regelmatig voor dat we een rapport maken op basis van een test database. Ter 

afronding zullen we dan het rapport weer moeten koppelen aan de productiedatabase. 

 

Om dat te doen gaan we naar het menu Database: 

 

 

 

▪ We kiezen daar Set Datasource Location 

In het bovenste scherm verschijnen de bestaande connectie en de gebruikte tabellen. In het onderste 

scherm kunnen we een andere connectie kiezen. 

 

We klikken dan eerst op de bovenste connectie en vervolgens op de onderste en klikken dan op Update. 

De bovenste connectie moet dan vervangen zijn door de onderste.  

 

In de praktijk kan het gebeuren dat er verschillen zijn tussen de twee databases. Tabellen en velden 

kunnen in namen van elkaar verschillen. We zullen dan moeten proberen detail voor detail te updaten. 

3.2 Opgave 

▪ Maak twee mappen op bijvoorbeeld de C schijf met de namen Test en Productie 

▪ Plaats in beide mappen de database Noordenwind 

▪ Maak een simpel rapportje op basis van de tabel Klanten op basis van de database Noordenwind 

uit de map Test 

▪ Sla het rapport op als oefening0401.rpt 

▪ Update nu de databasekoppeling naar de database Noordenwind uit de map productie. 

▪ Sla het rapport opnieuw op. 

4 Een tweede rapport in de vorm van een sjabloon 

4.1 Inleiding 

Crystal Reports kent geen echte sjablonen zoals we die bijvoorbeeld in MS Word hebben. Een alternatief 

is een voorbeeldbestand dat we zelf maken.  

 

Wat moeten we in dit voorbeeldbestand vastleggen? We kunnen denken aan de volgende punten: 

 

▪ Printer: is per rapport vast te leggen 

▪ De marges: afhankelijk van de gekozen printer  

▪ Pagina-instelling: staand of liggend 

▪ Summary info: basisgegevens over auteur etc. 

▪ Ingevoegde Special Fields: bijvoorbeeld Titel, commentaar en auteur 

▪ Paginanummering 

▪ Pad en naam onderaan in de Report Footer 

▪ Vermelden eventuele filters via Special Field Record Selection Formula 

▪ Plaatje en/of logo in de Report Header 

▪ Printen onderdrukken als er geen records zijn (vooral belangrijk bij subrapporten; ook voor alle 

rapporten in te stellen) 

▪ Data wel of niet bij het rapport bewaren (ook voor alle rapporten in te stellen) 

▪ Read only maken (het rapport kan nu niet meer per ongeluk overschreven worden) 

 

We gaan nu een sjabloon rapport maken en de genoemde stappen één voor één langs lopen.  

 

▪ Maak een nieuw rapport 

▪ Kies voor een ‘blank report’  

▪ sluit de Data explorer (een koppeling hebben we bij een sjabloon nog even niet nodig) 

4.2 Printer 

▪ Kies File  Print … 

De getoonde printer wordt in het rapport opgeslagen. 

 

▪ Kies eventueel een andere 

▪ Sla het rapport op als Sjabloon.rpt 

 

We kiezen bij voorkeur de printer waarop we ook daadwerkelijk gaan afdrukken! 

4.3 De marges 

▪ Kies File  Page setup 

▪ Stel alle marges bijvoorbeeld in op 2 cm 

4.4 Pagina-instelling: staand of liggend 

 

▪ Kies File  Print … 

▪ Kies voor Portrait (staand) of Landscape (liggend) 

 

Ook zaken als papierformaat en papierinvoer zijn hier in te stellen 

4.5 Summary info 

▪ Kies File  Summary info 

▪ Vul alle velden even in 

 

Een aantal van deze velden kunnen we als Special fields in het rapport plaatsen. 

4.6 Special fields 

▪ Klik in de Field explorer op de + voor Special Fields; we krijgen dan onderstaand scherm 

Hebben we de Field Explorer per ongeluk uit gezet, dan kunnen we die aanzetten via de knop  met 

als hulptekst Insert Fields 

 

 

 

▪ Zet van deze velden de Report Title in de band Report Header 

▪ Zet de velden File Author en Report Comments in de Report Footer 

▪ Denk er aan het sjabloon tussentijds op te slaan 

 

Van deze Special Fields nemen we er gelijk nog een aantal mee: 

 

▪ Zet het veld Page N of M in de Page Footer rechts 

▪ Zet het veld File Path and Name ook in de Page Footer maar dan links 

▪ Zet de velden Record Selection Formula en Group Selection Formula in de Report Header; deze 

tonen de eventueel door ons gemaakte selectie 

4.7 Plaatje en/of logo 

▪ Klik op de knop Insert Picture 

▪ Voor het invoegen van een plaatje klikken we nu op de derde knop van links 

▪ Zoek in het getoonde venster naar een map die plaatjes bevat en open het plaatje 

 

Aan de muiscursor hangt nu een rechthoek. 

 

▪ Plaats deze in de Report Header en klik 

4.8 Extra opties 

▪ Kies File  Report Options 

 

 

 

▪ Zet hier Save Data With Report uit of aan 

▪ Zet Suppress Print If No Records en Show All Headers On Drill Down (uitleg volgt later) beide aan 

▪ Sla de eindversie van het sjabloon op en kies in de verkenner de Properties (via de rechter 

muisknop) om Read Only op aan te zetten (dit kan ook via de Report Options in Crystal Reports 

zelf) 

4.9 Opgaven 

▪ Maak een nieuw rapport zonder een tabel of tabellen te kiezen 

▫ Kies een printer 

▫ Stel alle marges in op 2 centimeter   

▫ Kies als pagina-instelling staand 

▫ Vul de Summary info in 

▫ Voeg de Special Fields, bijvoorbeeld Titel, commentaar en auteur, toe aan het rapport 

▫ Plaats pagenumber in de page footer 

▫ Voeg pad en naam toe aan de report footer 

▫ Plaats een plaatje en/of logo in de page header 

▫ Geef aan dat de data niet met het rapport opgeslagen moeten worden 

▫ Sla het rapport op als sjabloonstaand.rpt 

5 Terug naar het stappenplan 

We weten nu hoe we een simpel rapport moeten maken. We hebben een bestaande Data Source Name 

die we steeds opnieuw kunnen gebruiken voor het selecteren van tabellen. Het wordt nu tijd voor 

ingewikkelder zaken. 

 

We brengen eerst weer even het vaste stramien in herinnering: 

 

1. Welke informatie wordt er gevraagd? 

2. Welke database(s) moeten we gebruiken? 

3. Welke tabellen/queries daaruit hebben we nodig? 

4. Welke relaties liggen er tussen tabellen? 

5. Welke velden selecteren we? 

6. Welke gegevens moeten we afleiden uit welke velden? 

7. Welke restricties passen we toe op welke velden of afleidingen? 

8. Hoe groeperen we? 

9. Welke sorteervolgorde kiezen we? 

10. Welke cumulatieve bewerkingen (totalen, gemiddelden, aantallen etc.) voor welke groepen van 

gegevens? 

11. Hoe maken we het rapport op? 

 

In de volgende hoofdstukken gaan we dit stramien vanaf stap 6 verder toelichten. 

6 Formula fields 

6.1 Inleiding 

Vaak is informatie niet direct voorhanden maar moeten we afleiden uit één of meer velden. 

Omzetgegevens zullen zelden direct ergens in een tabel staan; we zullen deze gegevens veelal moeten 

afleiden door bijvoorbeeld de velden prijs en verkochte aantallen met elkaar te vermenigvuldigen.  

 

In Crystal Reports gebruiken we de Formula fields om dergelijke afleidingen te maken. Formula Fields 

bieden verder tal van extra functies om gegevens om te zetten.  

6.2 Werkwijze 

Hoe maken we dergelijke velden? Stel we willen alle plaatsnamen groeperen op de eerste letter. Dan 

hebben we een Formula Field nodig dat de eerste letter uit een veld Plaats haalt. We gaan als volgt te 

werk:  

 

▪ in de Field Explorer klikken we Formula fields en dan op de knop New 

 

  

 

▪ In het volgende scherm geven we het veld een naam: 

 

 

 

▪ Klik op OK  

 

We komen dan in het volgende scherm: 

In het vak rechtsonder stellen we onze Formula samen: 

 

▪ We kiezen uit het middelste vak uit de rubriek Strings de functie Left(str, length).  

 

In het onderste scherm verschijnt:  

 

Left ( , ).  

 

▪ We plaatsen de cursor voor de komma en kiezen uit het vak linksboven het veld Klanten.Plaats.  

▪ Achter de komma plaatsen we een 1.  

 

De uiteindelijke formula ziet er zo uit:  

 

 Left ({Klanten.Plaats}, 1).  

 

▪ Om de formula te bewaren, klikken we op de knop Save and Close. 

 

We kunnen nu de formula in ons rapport plaatsen. Als we nu groeperen (zie Hoofdstuk Fout! V

erwijzingsbron niet gevonden.) op Formula Field Eersteletter, dan worden alle plaatsen op die manier 

gegroepeerd. 

6.3 Losse voorbeelden 

We gaan nu de Formula Fields verder toelichten aan de hand van een aantal voorbeelden. 

 

a. Leeftijd afleiden uit een geboortedatum 

Crystal Reports kent weliswaar de functie Datediff maar anders dan in Excel bijvoorbeeld geeft deze 

alleen het verschil tussen de jaren en houdt geen rekening met de feitelijke datums.   

 

We hebben daarom de volgende formule nodig:   

 

 Int((CurrentDateTime-{Werknemers.Geboortedatum})/365). 

 

Helemaal correct is dit niet. Dat zou als volgt moeten: 

 if month(currentdate)> month ({Werknemers.Geboortedatum}) then 

    "leeftijd: "  & year(currentdate) - year ({Werknemers.Geboortedatum}) 

 else 

    if month(currentdate)< month({Werknemers.Geboortedatum}) then 

        "leeftijd: "  & year(currentdate) - year 

({Werknemers.Geboortedatum}) - 1 

    else 

        if day({Werknemers.Geboortedatum})< day(currentdate) then 

            "leeftijd: "  & year(currentdate) - year 

({Werknemers.Geboortedatum}) -1 

        else            

            "leeftijd: "  & year(currentdate) - year 

({Werknemers.Geboortedatum}) 

 

b. Leeftijd plus tekst 

Combineren we een stuk tekst met een getal dan zullen we het getal om moeten zetten naar tekst met 

behulp van de ToText functie. Anders geeft Crystal Reports een foutmelding, iets als: 

 

 

 

De uiteindelijke formula ziet er dan zo uit: 

 

 "Uw leeftijd is:" + ToText(Int((CurrentDateTime-

{Werknemers.Geboortedatum})/365)) 

 

c. Berekening 

We kunnen ook echte berekeningen maken. Om de netto omzet per orderregel uit te rekenen:  

{Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid}*(1-{Orderinformatie.Korting}) 

 

d. Teksten uit een tekst vissen 

Hoe bepalen we of het woord dam voorkomt in een ander woord? De volgende functie geeft aan op 

welke positie het woord dam eventueel begint: 

 

 Instr('amsterdam', 'dam') 

 

Het resultaat is 7. Dus dam begint op positie 7. Komt het niet voor dan is het resultaat 0!  

 

e. Control structures 

Het kan ook ingewikkelder. Stel iemands geboortedatum is onbekend. We kunnen dan de berekening 

niet maken. Dan kunnen we dat zo oplossen: 

 

 if not isnull({Werknemers.Geboortedatum})   then 

    "Uw leeftijd is: " + ToText(Int((CurrentDateTime-

{Werknemers.Geboortedatum})/365)) 

 else 

    "Uw leeftijd is onbekend" 

 

Voortbordurend op voorbeeld d.: 

 If Instr({klanten.plaats}, “dam”) > 0 then  

  {klanten.plaats} 

 else 

  “er komt geen dam in voor” 

 

Nog een voorbeeld. We willen het kwartaal uit de maand afleiden met een zogenaamde Select Case 

constructie. Select case geniet de voorkeur boven een contructie met geneste ifs als er meerdere opties 

zijn en als het aantal opties makkelijk uitbreidbaar moet zijn. 

 

We kunnen dat dan zo formuleren: 

 

 totext(year({Werknemers.Geboortedatum}),0,"") + " " +  

 select month({Werknemers.Geboortedatum}) 

    case 1 to 3: "1ste kwartaal" 

    case 4 to 6: "2de kwartaal" 

    case 7 to 9: "3de kwartaal" 

    case 10 to 12: "4de kwartaal" 

 

De totext functie gebruiken we hier in een bepaalde variant. De 0 erachter geeft aan dat we geen cijfers 

achter de komma willen; de dubbele aanhalingstekens met spatie ertussen zorgen ervoor dat er geen 

punt voor de duizendtallen wordt gebruikt. De functie totext sec had namelijk een getal met de vorm 

1.999,00 opgeleverd voor het jaar 

 

We hadden ook kunnen gebruiken  

 

 totext(year({Werknemers.Geboortedatum}),"#")  

 

of  

 

 totext(year({Werknemers.Geboortedatum}),"0000") 

6.4 Opgaven 

▪ Maak een rapport op basis van de tabel Orderinformatie 

▫ plaats daarvoor de velden order-id, prijs per eenheid, hoeveelheid en korting in de detailsectie 

▫ maak Formula fields die berekenen wat de Bruto en de Netto omzet (dus inclusief korting) per 

orderregel is 

Bruto: {Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid}) 

Netto: {@bruto}*(1-{ Orderinformatie.Korting}) 

▫ plaats deze eveneens in de detailsectie 

▫ sla het rapport op als oefening0601.rpt 

▪ Maak een rapport op basis van de tabel producten 

▫ plaats daarvoor de velden voorradig, in bestelling en prijs per eenheid in de detailsectie 

▫ maak Formula fields die berekenen wat de voorraadwaarde per product is, wat de waarde is 

van de bestelling  en van het totaal van Voorradig en In bestelling 

Voorraadwaarde: {Producten.Prijs per eenheid}*{Producten.Voorradig} 

Waardebestelling: {Producten.Prijs per eenheid}*{Producten.In bestelling} 

Totaal voorradig + bestelling: {@voorraadwaarde} +{ @waardebestelling} 

▫ plaats deze eveneens in de detail sectie 

▫ sla het rapport op als oefening0602.rpt 

▪ Maak een rapport op basis van de tabel werknemers 

▫ plaats de velden achternaam en geboortedatum in de details sectie 

▫ maak een veld dat de leeftijd berekent 

leeftijd:  

int((currentdate - {Werknemers.Geboortedatum})/365.23) 

//functie int kapt af naar beneden 

▫ plaats dit eveneens in de detailsectie 

▫ sla het rapport op als oefening0603.rpt 

▪ Maak een rapport op basis van de tabel Orders 

▫ plaats de velden order-id en orderdatum in de detailsectie 

▫ maak een formula field dat de orderdatums omzet naar halfjaarlijkse perioden met een Select 

Case (kan ook met If Then Else) 

halfjaar:  

 Select month({Orders.Orderdatum}) 

  case is < 7 : "eerste helft" 

  default: "tweede helft" 

 

of: 

 

 If  month({Orders.Orderdatum}) < 7 then 

   "eerste helft" 

 Else 

   "tweede helft" 

 

▫ plaats dit eveneens in de detail sectie 

▫ sla het rapport op als oefening0604.rpt 

 

▪ Maak een rapport op basis van de tabel Producten 

 

▫ plaats de velden productnaam en hoeveelheid per eenheid in de detailsectie 

▫ maak een Formula Field dat met behulp van de Instr functie bij alle records met het woord fles 

in de kolom Hoeveel per eenheid als uitkomst waar geeft en anders onwaar 

flesofniet: 

 

 if instr({Producten.Hoeveelheid per eenheid},"fles") > 0 then 

    "waar" 

 else 

    "onwaar" 

 

▫ plaats dit eveneens in de detailsectie 

▫ sla het rapport op als oefening0605.rpt 

7 Selecteren van records 

7.1 Select Expert 

De Select Expert gebruiken we om gegevens te filteren. 

 

▪ Maak eerst een simpel rapport op basis van de tabel klanten met de velden Klantnummer, bedrijf en 

contactpersoon.  

▪ Klik vervolgens op Print Preview 

 

Op de tabel Klanten gaan we nu een aantal selecties toepassen. 

 

▪ We klikken nu het veld Contactpersoon aan 

▪ En klikken dan op de knop  Select Expert 

 

We krijgen een volgend scherm: 

 

 

 

 

Let op:  

Het aantal getoonde waarden in het rechter pull down vak is maximaal 500 verschillende. 

 

Let op:  

Als je geen veld uit het rapport geselecteerd hebt, verschijnt eerst nog dit scherm: 

Hieruit moet je dan eerst nog een veld voor selectie kiezen! 

 

 

▪ Kies in het eerste popup scherm voor is equal to en in het tweede Ann Devon; het volgende scherm 

verschijnt (alleen als we in de preview weergave zitten) 

 

 

 

▪ Klik op Refresh Data 

 

Het resultaat is dat we alleen nog de gegevens van de gekozen dame zien. 

 

▪ We klikken opnieuw op de Select Expert 

▪ En klikken op New 

 

 

 

▪ Hier kiezen we voor Plaats 

 

 

Let op: de bovenliggende velden maken deel uit van het rapport; onderin zien we alle velden van de 

tabel! 

 

 

▪ We kiezen dan voor Is Equal To en Berlijn 

 

Als we dan op Refresh klikken, krijgen we helemaal niets. Hoe kan dit? Dubbele voorwaarden worden 

automatisch And voorwaarden. Het systeem gaat dus op zoek naar iemand die Ann Devon heet en in 

Berlijn woont. En die is er niet. 

 

▪ Klik opnieuw op de Select Expert  

▪ Verander het woord and in or  

▪ En ververs de data weer 

 

En nu krijgen we er twee! 

 

Er zijn ook nog verschillende andere selectiewijzen zoals is one of en is between. 

 

 

 

 

 

Let op:  

Via Special fields  Record Selection Formula kunnen we zichtbaar maken welke selectie we hebben 

uitgevoerd.  

Zorg er bij navolgende oefeningen steeds voor dat dit Special Field in de Report Header staat! 

7.2 Filteren op tekst uit een memoveld 

Het instellen van een filter op een veld van het type memo lukt normaal gesproken niet in Crystal 

Reports. Er is evenwel een workaround. Maak voor het volgende voorbeeld gebruik van een tabel dat 

een veld van het type memo bevat. 

 

▪ maak een parameter met bijvoorbeeld de naam FILTER, type STRING 

▪ maak een Formula Field met bijvoorbeeld de naam FILTER 

▪ plaats daarin de volgende code: {tabel.veldnaam} like "*" & {?filter} & "*"  
▪ maak een filter op dit Formula Field en kies IS TRUE 

 
Het memoveld zal dan gefilterd worden op de in de parameter opgegeven tekst 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 35 

7.3 Selectie op groepsresultaten 

Het is ook mogelijk om op groepsresultaten te selecteren. Zie scherm: 

 

 

 

Bij de formula editor zien we dan allen de formulas die betrekking hebben op de groepsselectie. Kiezen 

we voor record selection dan zien we de andere formulas. Dit komt pas aan de orde bij het hoofdstuk 

groeperen. 

 

7.4 Opgaven 

▪ Maak een rapport op basis van de tabel werknemers  

▫ plaats de velden achternaam en functie in de detailsectie 

▫ Maak een selectie op het veld functie en selecteer alleen de vertegenwoordigers  

▫ sla het rapport op als oefening0701.rpt 

 

▪ Maak een rapport van de tabellen producten en categorieën 

▫ plaats de velden productnaam, categorienaam en prijs per eenheid in de detailsectie 

▫ selecteer alleen de categorie dranken en de producten met prijs per eenheid boven de 20 (and 

voorwaarde) 

▫ sla het rapport op als oefening0702.rpt 

 

▪ Maak een rapport van de tabel producten  

▫ plaats de velden productnaam en prijs per eenheid in de details sectie 

▫ selecteer de producten met een prijs per eenheid tussen de 10 en 20 (met optie Is Between) 

▫ sla het rapport op als oefening0703.rpt 

 

▪ Maak een rapport van de tabel klanten  

▫ plaats de velden bedrijf en plaats in de detailsectie 

▫ selecteer alleen de klanten uit de plaatsen Berlijn, Madrid en Barcelona (met optie Is one of) 

▫ sla het rapport op als oefening0704.rpt 

 

▪ Maak een rapport van de tabellen orders en verzenders 

▫ koppel de bestanden verzenders en orders op basis van de velden verzend-id – verzendwijze 

▫ plaats de velden order-id en orderdatum en bedrijf in de detailsectie 

▫ selecteer alleen de orders augustus 1996 (met optie Is Between tussen 1 augustus en 1 

september; let op de tijd!), met verzendbedrijf Federal Shipping en vrachtkosten boven de 100 

▫ sla het rapport op als oefening0705.rpt 

 

▪ Maak een rapport van de tabel Orders  

▫ plaats de velden order-id, orderdatum en vrachtkosten in de details sectie 

▫ selecteer alleen de orders uit 1996 of (verander and in de formula in or) de orders met minder 

dan 100 aan vrachtkosten 

▫ sla het rapport op als oefening0706.rpt 

 

▪ Maak een rapport van de tabel Producten 

▫ plaats de velden productnaam en hoeveelheid per eenheid in de detailsectie 

▫ selecteer alle records die in het veld Hoeveelheid per eenheid het woord fles bevatten 

▫ sla het rapport op als oefening0707.rpt 

 

▪ Maak een rapport van de tabel Orders  

▫ plaats de velden order-id en orderdatum in de detailsectie 

▫ selecteer met de Orderdatum alleen gegevens uit het derde kwartaal van elk jaar m.b.v. de 

datepart  functie, formula field kwartaal:  

 

 datepart("q",{Orders.orderdatum}) 

 

▫ sla het rapport op als oefening0708.rpt 

 

▪ Maak een rapport van de tabel Werknemers 

▫ plaats de velden achternaam en geboortedatum in de detailsectie 

▫ maak een formula field dat de leeftijd berekent  

 

 int((currentdate - {Werknemers.Geboortedatum})/365.23) 

 

▫ plaats dit ook in de detailsectie 

▫ selecteer alleen de werknemers met een leeftijd hoger dan 45 

▫ sla het rapport op als oefening0709.rpt 

8 Groeperen en sorteren 

8.1 Groeperen 

8.1.1 Algemeen 

Groeperen betekent letterlijk velden met dezelfde inhoud bij elkaar nemen. Per groep kunnen we 

vervolgens iets gaan uitrekenenen: een gemiddelde, een minimum etc.  

Hoe werkt dit in Crystal Reports? 

 

▪ Maak eerst een simpel rapport van de tabel klanten met de velden Klantnummer, bedrijf en 

contactpersoon.  

▪ Klik vervolgens op Print Preview 

▪ Klik vervolgens op de Insert a Group knop  

 

 

 

 

Ook vinden we op een van de werkbalken een extra knop Group Expert die ons via de knop Options 

ook weer naar het bovenstaande scherm brengt. 

▪ Kies uit de tabelvelden voor Land en klik op OK;  

 

We krijgen het volgende beeld: 

 

 

 

De gegroepeerde velden zijn automatisch in de groepsheaders geplaatst. 

8.1.2 Customize Group Name Field 

We hadden ook kunnen kiezen voor de optie Customize Group Name Field.  

Wanneer gebruiken we dat? Stel je voor we willen op naam groeperen. We lopen dan het risico dat er 

dubbele namen voorkomen (bijv. Jan Jansen). We kiezen er dan voor om op de code van de persoon te 

groeperen en de naam te tonen. We kunnen dan bijvoorbeeld het veld Achternaam kiezen of een 

Formula gebruiken om Achternaam en Voornaam weer te geven. 

Een ander voorbeeld is het groeperen op datum waarbij we kiezen voor kwartaal. Crystal Reports geeft 

dit dan weer als bijvoorbeeld 7-96, oftewel de eerste maand van het kwartaal plus het jaar. Mooier is het 

dan te kiezen voor Use a Formula as Group Name, met een  formula die op basis van het 

maandnummer een kwartaal toont: 

 

 totext(year({Orders.Orderdatum}),0,"") + " " +  

 select month({Orders.Orderdatum }) 

    case 1 to 3: "1ste kwartaal" 

    case 4 to 6: "2de kwartaal" 

    case 7 to 9: "3de kwartaal" 

    case 10 to 12: "4de kwartaal" 

 

of: 

 

 totext(year({Orders.Orderdatum}),0,"") + " kwartaal " 

+totext(datepart("q", {Orders.Orderdatum}),0) 

8.1.3 Groeperen op datums 

Wanneer we een datumveld kiezen om op te groeperen, krijgen we een speciaal scherm met extra 

mogelijkheden: 

8.1.4 Groepering veranderen 

Om een groepering te veranderen klikken we met de rechter muisknop op de Group Header of Footer, 

direct links van het rapport. 

 

▪ Verander de groepering van Land naar Plaats en bekijk het resultaat 

 

We kunnen natuurlijk ook een groep in een groep maken. 

 

▪ Verander Plaats weer terug in Land 

▪ Voeg een tweede groep in en kies nu voor Provincie 

▪ Bekijk het resultaat 

 

We kunnen de volgorde van de groeperingen veranderen met behulp van de muis. We trekken dan de 

ene groep buiten de andere groep of omgekeerd 

8.2 Groeperen met Formula Fields 

We kunnen ook groeperen op een deel van velden of omgewerkte velden, met behulp van Formula 

Fields. Een voorbeeld.  

 

▪ Maak een blank report op basis van de tabel klanten 

▪ Maak via de Field Explorer een Formula Field met de naam pc2 

 

 

 

Het volgende scherm verschijnt: 

▪ Typ de formule: Left ({Klanten.Postcode}, 2) 

▪ Klik op de knop Save and Close 

▪ Groepeer vervolgens op het Formula Field 

 

 

8.3 Groeperen met Specified Order 

Crystal Reports heeft ook een bijzondere manier van groeperen: groeperen door zelf de inhoud van 

velden in te delen in specifieke groepen. Bijvoorbeeld Canada, USA en Brazilië vormen de groep Grote 

Landen. 

 

▪ Maak een blank report op basis van de tabel klanten 

▪ Groepeer op Land 

▪ Kies voor specified order  

Het volgende scherm verschijnt: 

 

 

 

▪ Klik hier op New 

 

Het volgende scherm verschijnt: 

▪ Bedenk een naam voor een groep, bijvoorbeeld Grote Landen 

▪ Kies Is one of  

▪ Voeg een aantal landen toe  

 

 

 

▪ Klik op OK 

▪ Maak vervolgens op vergelijkbare wijze nog een paar groepen  

▪ Klik in bovenstaande scherm opnieuw op New  

▪ Herhaal de vorige stappen 

▪ Bekijk het eindresultaat 

8.4 Sorteren 

Een groep is automatisch gesorteerd. Maar wat als we binnen een groep verder willen sorteren?  

▪ Dan drukken we op de knop Sort Order  voor het volgende scherm: 

 

 

 

▪ Binnen land en provincie sorteren we op Bedrijf 

8.5 Paginering 

We kunnen elke groep op een nieuwe pagina laten beginnen. We doen dat met de Section Expert: 

 

 

 

En 

 

 

 

We hebben dan twee mogelijkheden. We kiezen bij de Group Header de mogelijkheid New Page 

Before of bij de Group Footer de mogelijkheid New Page After.  

 

Vanaf versie 2008 is deze mogelijkheid ook verlegd naar de groep zelf. 

Let op:  

Beide mogelijkheden in de Section Expert leveren een probleempje op: of een lege pagina vooraan of 

ééntje achteraan. Dit voorkomen we door niet de mogelijkheid zelf aan te vinden maar op de knop 

erachter te klikken en daar in te vullen resp. Not OnFirstRecord of Not OnLastRecord (beide functies zijn 

te vinden bij de categorie Print State). 

 

 

We hebben ook de mogelijkheid om de paginanummering te manipuleren via Reset page number after, 

via de Section Expert: 

We kunnen er ook voor zorgen dat een groep bij elkaar blijft op één pagina en als dat niet mogelijk is, 

dat de groepskop herhaald wordt op elke pagina: 

 

 

 

Als we kiezen voor Repeat Group Header On Each Page onstaat er wel een probleem met het resetten 

van de paginanummering per groep. Het paginanummer wordt dan op elke pagina 1. Dit is te 

ondervangen door een tweede identieke groep te maken en de paginanummering voor die groep te 

resetten. 

 

8.6 Opgaven 

▪ Maak een rapport met de tabel leveranciers 

▫ plaats het veld bedrijf in de detailsectie 

▫ groepeer op land en daarbinnen op plaats 

▫ sla het rapport op als oefening0801.rpt 

 

▪ Maak een rapport met de tabellen Leveranciers en Producten 

▫ plaats het veld productnaam in de detailsectie 

▫ groepeer eerst op land  

▫ groepeer vervolgens binnen land op bedrijf (de volgorde van groeperen bepaalt de hierarchie; 

anders moeten we ze achteraf omdraaien) 

▫ sla het rapport op als oefening0802.rpt 

 

▪ Maak een rapport met de tabel klanten 

▫ plaats de velden bedrijf, adres en plaats in de detailsectie 

▫ maak een formula field eersteletter dat de eerste letter uit de plaatsnaam haalt; 

 

left({klanten.plaats},1) 

 

▫ groepeer het rapport op formula field eersteletter. 

▫ sla het rapport op als oefening0803.rpt. 

 

▪ Maak een rapport met de tabel Klanten. 

▫ plaats de velden land en bedrijf in de detailsectie. 

▫ groepeer deze op zelf gespecificeerde groepen met de namen Grote landen, middelgrote 

landen en kleine landen. 

 

 

 

▫ sorteer per groep op land hierop (Let op: dit kan alleen door eerst een Formula Field met als 

inhoud {Klanten.Land} te maken en hierop te sorteren; we zullen later uitleggen waarom). 

▫ sla het rapport op als oefening0804.rpt. 

 

▪ Maak een rapport met de tabel Orders. 

▫ plaats de velden orderdatum en order-id in de detailsectie. 

▫ groepeer per kwartaal. 

 

 

 

▫ laat via onderstaande scherm met een formula wel het kwartaal op de juiste manier zien,  

bijvoorbeeld:  

totext(year({Orders.Orderdatum}),0,"") + " kwartaal " 

+totext(datepart("q", {Orders.Orderdatum}),0) 

 

 

 

▫ sla het rapport op als oefening0805.rpt. 

 

▪ Maak een rapport met de tabellen werknemers en orders 

▫ plaats de velden order-id en orderdatum in de detailsectie. 

▫ groepeer op de code van de werknemers. 

▫ maar toon Achternaam en voornaam via Customize Group Name Field (zie scherm hiervoor). 

▫ voeg paginanummering toe via Special Fields uit de Field Explorer. 

 

 

 

▫ laat via de Section Expert voor elke groep de paginanummering weer bij 1 beginnen. 

▫ sla het rapport op als oefening0806.rpt. 

9 Totalen, aantallen, gemiddelden etc. 

9.1 Voorbeelden 

Een totaal, een aantal, een gemiddelde etc. berekenen we altijd over een groep, waarbij de grootste 

groep het geheel is. We hebben het dan over het generaal totaal. 

 

▪ We maken weer een rapport gebaseerd op de tabel klanten met de velden Klantnummer, Bedrijf en 

Contactpersoon en groeperen op Land 

 

Nu kunnen we accumuleren op twee verschillende niveaus: geheel en per land. 

 

▪ Klik in de detailsectie op het veld Contactpersoon 

▪ Klik op de rechter muisknop 

 

 

 

▪ Kies voor Grand Total 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 53 

 

 

▪ Kies voor Count en druk op OK 

 

In de Design mode zien we nu een totaal staan in de Report Footer. Dat is de plek waar een Grand Total 

automatisch geplaatst wordt door Crystal Reports. Het is evenwel niet noodzakelijk dat het uitgerekend 

daar staat.  

 

▪ We kunnen het veld zelf kopiëren naar de Report Header 

 

En zien dat het voor de uitkomst niet uitmaakt. We kunnen zelfs nog een stap verder gaan. Afhankelijk 

van de plek binnen de structuur van het rapport krijgt de formule een andere uitkomst: 

 

➢ Report Header en Footer: Grand Total 

➢ Group Footer en Header: subtotaal 

 

▪ Probeer het voorgaande uit door het Grand Total te kopiëren naar de Group Footer en Header 

We kunnen dan eventueel de detailsectie verbergen: 

 

▪ Klik rechts op de Detail sectie in de marge 

 

 

 

▪ Klik op Suppress (No-Drill Down) 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 55 

 

Het resultaat ziet er zo uit: 

 

 

 

 

 

Let op: 

Het kan zelfs ook de andere kant uit. We kunnen een element uit de detailsectie ook kopiëren naar de 

Group Footer en Header; we krijgen dan respectievelijk het laatste en eerste lid van de groep te zien. 

 

 

▪ Probeer dit uit en doe het zelfde door een lid van de detailsectie te kopiëren naar de Report Header 

en Footer 

9.2 Opgaven 

▪ Maak een rapport op basis van de tabellen producten en categorieën 

▫ maak formula fields die de voorraadwaarde en de waarde van de bestellingen berekenen: 

 

 {Producten.Prijs per eenheid}*{Producten.Voorradig} 

 {Producten.Prijs per eenheid}*{Producten.In bestelling} 

 

▫ plaats velden productnaam, voorradig, in bestelling en de formula fields voorraadwaarde en 

waardebstelling in de detailsectie 

▫ groepeer op categorienaam 

▫ sommeer vervolgens de waarde van de voorraad en van de bestellingen per categorie 

▫ sla het rapport op als oefening0901.rpt 

 

▪ Maak een rapport van de tabel werknemers  

▫ maak een formula field dat de leeftijd berekent 

 

int((currentdate - {Werknemers.Geboortedatum})/365.23) 

▫ plaats de velden achternaam, geboortedatum en leeftijd in de detailsectie 

▫ groepeer per beleefdheidstitel 

▫ bereken de gemiddelde leeftijd per beleefdheidstitel! 

▫ sla het rapport op als oefening0902.rpt 

 

▪ Maak een rapport van de tabellen Klanten, Orders en Orderinformatie 

▫ maak formula fields die de bruto en netto omzet berekenen 

 

 Bruto: {Orderinformatie.Prijs per 

eenheid}*{Orderinformatie.Hoeveelheid} 

 Netto: {@bruto}*(1-{ Orderinformatie.Korting} 

 

▫ plaats de velden order-id, orderdatum en bruto- en netto omzet in de detailsectie 

▫ groepeer op bedrijf 

▫ bereken de totale bruto- en netto omzet per klant 

▫ sla het rapport op als oefening0903.rpt 

 

▪ Maak een rapport van de tabellen Producten, Orders en Orderinformatie 

▫ maak formula fields die de bruto en netto omzet berekenen 

 

 Bruto: {Orderinformatie.Prijs per 

eenheid}*{Orderinformatie.Hoeveelheid} 

 Netto: {@bruto}*(1-{ Orderinformatie.Korting} 

 

▫ plaats de velden order-id, orderdatum en bruto- en netto omzet in de detailsectie 

▫ groepeer op productnaam 

▫ bereken de omzet per product per kwartaal  

▫ sla het rapport op als oefening0904.rpt 

10 Opmaak: what you get is what is visible! 

10.1 Simpele opmaak 

De meeste opmaak is betrekkelijk simpel tenzij we conditionele opmaak gaan gebruiken. 

 

Zo ontdekken we allerlei knoppen die we kennen uit andere pakketten en die eender werken: vet, 

cursief, onderstrepen enzovoorts. 

 

Voor de opmaak van getallen, datums kunnen we ook een beroep doen op de rechter muisknop of de 

Tool Bars. Bij klikken op een veld met de rechter muisknop krijgen we via de keuze Format Field het 

volgende scherm: 

 

 

 

Ook hier vinden we tal van mogelijkheden.  

10.2 Uitlijnen en de breedte van meer velden  aanpassen 

Als je een aantal velden wilt uitlijnen of van dezelfde breedte wilt maken, kun je dat in één keer doen 

door ze allemaal te selecteren. 

 

Vervolgens is het van belang op welk veld je staat als je op de rechter muisknop klikt. Het gekozen veld 

is namelijk bepalend voor de uitlijning en de breedte. 

10.3 Conditionele opmaak 

Wat verstaan we daar onder? We kunnen bijvoorbeeld een detailsectie conditioneel wel of niet tonen. 

We kunnen de detailsectie alternerend laten kleuren: wit – grijs, wit grijs etc. Hoe doen we dit? Laten we 

met het laatste voorbeeld beginnen.  

 

▪ We beginnen met de Section Expert 

▪ Rechts klik op de muis voor de kantlijn 

▪ Kies de Section Expert 

▪ We komen dan in de Section Expert.  

▪ In het volgende dialoogvenster klikken we op Color: 

 

 

 

 

We komen dat in een scherm dat sterk lijkt op dat waarmee we de Formula fields konden maken.  

 

 

 

Er is echter een belangrijk verschil in uitkomst: een formula field levert een waarde op. In dit geval wordt 

een kleur terugverwacht. Vaak ook wordt hier een True of False terugverwacht. Een True om wel iets te 

doen, een False om niet iets te doen. 

▪ We tikken onderin de volgende code:  

 

 if Remainder(RecordNumber,2)=0 then crSilver else NoColor 

 

Vrij vertaald staat hier: als de rest van de deling van het recordnummer 0 is, dan de kleur Zilver, anders 

geen kleur. De geretourneerde kleur wordt vervolgens gegeven aan de detailsectie! 

 

Het zelfde effect bereiken we ook met onderstaande Formula 

 

 if Recordnumber mod 2 = 1 then 

  crSilver 

 else 

  NoColor 

 

Houden we niet van typen, dan kan het ook bijna volledig menugestuurd: 

▪ Klik onder Operators op Control structures 

▪ Dubbelklik op if x then y else z 

▪ Plaats de cursor na de if  

▪ Klik onder functions op math 

▪ Dubbelklik op Remainder 

▪ Plaats de cursor achter het haakje ( 

▪ Sluit math weer met een enkele klik 

▪ Klik vervolgens onder functions op print state 

▪ Dubbelklik op RecordNumber 

▪ Zet achter de komma het getal 2 en achter het haakje ) = 0 

▪ Plaats de cursor achter then 

▪ Sluit print state weer met een enkele klik 

▪ Klik onder fucntions op Color constants 

▪ Dubbelklik op Silver 

▪ Plaats de cursor achter else 

▪ Dubbelklik op NoColor 

 

We slaan de code op met de knop linksboven. Let erop dat dan de knop achter Background color rood 

geworden is! 

We krijgen dan het volgende beeld: 

 

 

 

Nog een voorbeeld. We willen de klant niet tonen als hij uit Madrid komt.  

 

▪ We gaan weer naar de Section Expert. 

 

 

 

▪ We klikken dan op de knop achter Suppress.  

▪ We komen weer in het scherm waar we programmacode kunnen invoeren.   

▪ Maar we hebben nu een voorbeeld te pakken waarbij het programma een True of False verwacht. 

En bij True gaat het wel onderdrukken. 

 

▪ Als code tikken we in: {Klanten.Plaats}="Madrid".  

 

In het resultaat zullen we ontdekken dat de klanten uit Madrid niet langer getoond worden. 

Crystal Reports ondersteunt Boolean operators maar toont ze als: 

 

True of False, 0 of 1 enzovoorts 

 

Crystal Reports kan deze operators niet standaard tonen als een vierkantje resp. vierkantje met vinkje 

maar met wat creativiteit en Wingdings lukt dat wel. 

 

▪ Als je een Boolean veld in je rapport geplaatst hebt, kies je de Format Field (rechts klikken op het 

veld)  

▪ Op het tabblad Font kies je bij Font voor Wingdings 

 

 

 

▪ Op het tabblad Common klik je op de knop achter Display Strings op de knop: 

▪ Onder die knop maak je de formule: 

 

 if {Producten.Uit assortiment}= true then 

    chr(254) 

 else 

    chr(111) 

 

Het uiteindelijke resultaat wordt dan: 

10.4 Opgaven 

▪ Maak een nieuw rapport met de tabel producten 

▫ Plaats een aantal velden in de detailsectie 

▫ Laat de detailsectie om en om kleuren 

▫ sla het rapport op als oefening1001.rpt 

 

▪ Maak een nieuw rapport met de tabel producten 

▫ Plaats een aantal velden in de detailsectie 

▫ Maak nu na elk groepje van vijf een regel met een dikke lijn (gebruik hiervoor een dubbele 

detailsectie en facultatief onderdrukken met behulp van de formula:  

 

 Recordmumber mod 5<> 0 

 

 

 

▫ sla het rapport op als oefening1002.rpt 

 

▪ Maak een nieuw rapport met de tabel leveranciers  

o Zorg dat alleen de details met leveranciers uit Frankrijk een gele achtergrond krijgen 

met Formula: 

 

 if {Klanten.Land} = "Frankrijk" then 

    cryellow 

 else 

    nocolor 

 

▫ sla het rapport op als oefening1003.rpt 

 

▪ Maak een nieuw rapport met de tabel orders en orderinformatie 

▫ Alleen totale orderbedragen boven de 10.000,- moeten een rode achtergrond krijgen 

▫ Groepeer op order-id 

▫ Maak een formula field Bruto dat de omzet per regel berekend: 

 

 {Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid}  

▫ Plaats het veld in de detailsectie 

▫ Totaliseer het veld 

 

 

 

▫ Plaats het in de group header van Order-id 

▫ Rechts klik op het Summary field 

▫ Kies Format Field 

▫ Kies border 

▫ Klik op de knop achter background 

▫ Plaats daar de Formula: 

 

 if Sum ({@bruto}, {Orderinformatie.Order-id}) > 10000 then 

    crred 

 else 

    nocolor 

 

▫ sla het rapport op als oefening1004.rpt 

 

▪ Maak een nieuw rapport met de tabel klanten 

▫ waarbij de paginanummers beurtelings links en rechts staan 

▫ er zijn een aantal alternatieve oplossingen 

▫ één paginanummer paginabreed waarbij de uitlijning alternerend links of rechts is 

En als Formula: 

 

 if pagenumber mod 2 = 0 then 

    crLeftAligned 

 else 

    crRightAligned 

 

▫ twee pagina secties a en b met een pagenumber in de eerste links en de tweede rechts; deze 

onderdrukken we dan beurtelings 

▫ sla het rapport op als oefening1005.rpt 

 

▪ Maak een rapport met de tabel producten 

▫ Plaats een plaatje (bijvoorbeeld uit de Windows map) in de page header  

▫ Kies bij de Section Expert voor Underlay following sections 

▫ sla het rapport op als oefening1006.rpt 

11 Opslaan van een rapport in een ander formaat 

11.1 Inleiding 

Het is in Crystal Reports mogelijk een rapport op te slaan in verschillende formaten. We kiezen daarvoor 

File  Export  Export Report 

 

 

 

We krijgen dan het volgende scherm: 

 

 

 

En hier kunnen we te kust en te keur gaan. 

 

Bijzondere aandacht verdient hierbij: de optie Report Definition. Met deze optie is het mogelijk een 

gedetailleerde beschrijving van het rapport naar een tekstbestand te laten wegschrijven. Zie voorbeeld: 

 Crystal Report Professional v10.0 (32-bit) - Report Definition 

 

 1.0 File Information 

 

 Report File:  

 Version: 10.2 

 

 2.0 Record Sort Fields 

 

 3.0 Group Sort Fields 

 

 4.0 Formulas 

 

 4.1 Record Selection Formula 

 {Leveranciers.Land} = "Canada" 

 

 4.2 Group Selection Formula 

  

 

 4.3 Other Formulas 

 

 5.0 Sectional Information 

 

 5.1 Page Header Section 

 Visible, Keep Together 

 

 Bedrijf 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 

 

 Plaats 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 

 

 5.2 Page Footer Section 

 Visible, New Page After, Keep Together, Print At Bottom of Page 

 

 5.3 Report Header Section 

 Visible, New Page Before 

 

 5.4 Report Footer Section 

 Visible, New Page After 

 

 5.5 Group Header Section 

 Visible 

 

 GroupName ({Leveranciers.Land}) 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 

  

 5.6 Group Footer Section 

 Visible 

 

 5.7 Details Section 

 Visible 

 Subsection.1  

  Visible, Keep Together 

 

 {Leveranciers.Bedrijf} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 

 

 {Leveranciers.Plaats} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

11.2 Opgaven 

▪ Maak een rapport van de tabel Leveranciers  

▫ plaats de velden bedrijf en plaats in de detailsectie 

▫ sla dit op als een HTML bestand 

▫ bekijk het resultaat in de browser 

▫ sla het rapport op als oefening0301.rpt 

 

▪ Gebruik het vorige rapport  

▫ groepeer op Land 

▫ maak een selectie op Canada 

▫ maak een export van Report Definition naar een txt bestand op de desk top 

▫ bekijk dit tekstbestand 

▫ sla het rapport op als oefening0302.rpt 

▫ sla het tekstbestand op als oefening0302.txt 

12 Relatieschema 

 

 

 
▫  

13 Opmaak: what you get is what is visible! 

13.1 Simpele opmaak 

De meeste opmaak is betrekkelijk simpel tenzij we conditionele opmaak gaan gebruiken. 

 

Zo ontdekken we allerlei knoppen die we kennen uit andere pakketten en die eender werken: vet, 

cursief, onderstrepen enzovoorts. 

 

Voor de opmaak van getallen, datums kunnen we ook een beroep doen op de rechter muisknop of de 

Tool Bars. Bij klikken op een veld met de rechter muisknop krijgen we via de keuze Format Field het 

volgende scherm: 

 

 

 

Ook hier vinden we tal van mogelijkheden.  

13.2 Uitlijnen en de breedte van meer velden  aanpassen 

Als je een aantal velden wilt uitlijnen of van dezelfde breedte wilt maken, kun je dat in één keer doen 

door ze allemaal te selecteren. 

 

Vervolgens is het van belang op welk veld je staat als je op de rechter muisknop klikt. Het gekozen veld 

is namelijk bepalend voor de uitlijning en de breedte. 

13.3 Conditionele opmaak 

Wat verstaan we daar onder? We kunnen bijvoorbeeld een detailsectie conditioneel wel of niet tonen. 

We kunnen de detailsectie alternerend laten kleuren: wit – grijs, wit grijs etc. Hoe doen we dit? Laten we 

met het laatste voorbeeld beginnen.  

 

▪ We beginnen met de Section Expert 

▪ Rechts klik op de muis voor de kantlijn 

▪ Kies de Section Expert 

▪ We komen dan in de Section Expert.  

▪ In het volgende dialoogvenster klikken we op Color: 

 

 

 

 

We komen dat in een scherm dat sterk lijkt op dat waarmee we de Formula fields konden maken.  

 

 

 

Er is echter een belangrijk verschil in uitkomst: een formula field levert een waarde op. In dit geval wordt 

een kleur terugverwacht. Vaak ook wordt hier een True of False terugverwacht. Een True om wel iets te 

doen, een False om niet iets te doen. 

▪ We tikken onderin de volgende code:  

 

 if Remainder(RecordNumber,2)=0 then crSilver else NoColor 

 

Vrij vertaald staat hier: als de rest van de deling van het recordnummer 0 is, dan de kleur Zilver, anders 

geen kleur. De geretourneerde kleur wordt vervolgens gegeven aan de detailsectie! 

 

Het zelfde effect bereiken we ook met onderstaande Formula 

 

 if Recordnumber mod 2 = 1 then 

  crSilver 

 else 

  NoColor 

 

Houden we niet van typen, dan kan het ook bijna volledig menugestuurd: 

▪ Klik onder Operators op Control structures 

▪ Dubbelklik op if x then y else z 

▪ Plaats de cursor na de if  

▪ Klik onder functions op math 

▪ Dubbelklik op Remainder 

▪ Plaats de cursor achter het haakje ( 

▪ Sluit math weer met een enkele klik 

▪ Klik vervolgens onder functions op print state 

▪ Dubbelklik op RecordNumber 

▪ Zet achter de komma het getal 2 en achter het haakje ) = 0 

▪ Plaats de cursor achter then 

▪ Sluit print state weer met een enkele klik 

▪ Klik onder fucntions op Color constants 

▪ Dubbelklik op Silver 

▪ Plaats de cursor achter else 

▪ Dubbelklik op NoColor 

 

We slaan de code op met de knop linksboven. Let erop dat dan de knop achter Background color rood 

geworden is! 

We krijgen dan het volgende beeld: 

 

 

 

Nog een voorbeeld. We willen de klant niet tonen als hij uit Madrid komt.  

 

▪ We gaan weer naar de Section Expert. 

 

 

 

▪ We klikken dan op de knop achter Suppress.  

▪ We komen weer in het scherm waar we programmacode kunnen invoeren.   

▪ Maar we hebben nu een voorbeeld te pakken waarbij het programma een True of False verwacht. 

En bij True gaat het wel onderdrukken. 

 

▪ Als code tikken we in: {Klanten.Plaats}="Madrid".  

 

In het resultaat zullen we ontdekken dat de klanten uit Madrid niet langer getoond worden. 

Crystal Reports ondersteunt Boolean operators maar toont ze als: 

 

True of False, 0 of 1 enzovoorts 

 

Crystal Reports kan deze operators niet standaard tonen als een vierkantje resp. vierkantje met vinkje 

maar met wat creativiteit en Wingdings lukt dat wel. 

 

▪ Als je een Boolean veld in je rapport geplaatst hebt, kies je de Format Field (rechts klikken op het 

veld)  

▪ Op het tabblad Font kies je bij Font voor Wingdings 

 

 

 

▪ Op het tabblad Common klik je op de knop achter Display Strings op de knop: 

▪ Onder die knop maak je de formule: 

 

 if {Producten.Uit assortiment}= true then 

    chr(254) 

 else 

    chr(111) 

 

Het uiteindelijke resultaat wordt dan: 

13.4 Opgaven 

▪ Maak een nieuw rapport met de tabel producten 

▫ Plaats een aantal velden in de detailsectie 

▫ Laat de detailsectie om en om kleuren 

▫ sla het rapport op als oefening0201.rpt 

 

▪ Maak een nieuw rapport met de tabel producten 

▫ Plaats een aantal velden in de detailsectie 

▫ Maak nu na elk groepje van vijf een regel met een dikke lijn (gebruik hiervoor een dubbele 

detailsectie en facultatief onderdrukken met behulp van de formula:  

 

 Recordmumber mod 5<> 0 

 

 

 

▫ sla het rapport op als oefening0202.rpt 

 

▪ Maak een nieuw rapport met de tabel leveranciers  

▫ Zorg dat alleen de details met leveranciers uit Frankrijk een gele achtergrond krijgen met 

Formula: 

 

 if {Leveranciers.Land} = "Frankrijk" then 

    cryellow 

 else 

    nocolor 

 

▫ sla het rapport op als oefening0203.rpt 

 

▪ Maak een nieuw rapport met de tabel orders en orderinformatie 

▫ Alleen totale orderbedragen boven de 10.000,- moeten een rode achtergrond krijgen 

▫ Groepeer op order-id 

▫ Maak een formula field Bruto dat de omzet per regel berekend: 

 

 {Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid}  

▫ Plaats het veld in de detailsectie 

▫ Totaliseer het veld 

 

 

 

▫ Plaats het in de group header van Order-id 

▫ Rechts klik op het Summary field 

▫ Kies Format Field 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 82 

 

 

▫ Kies border 

▫ Klik op de knop achter background 

▫ Plaats daar de Formula: 

 

 if Sum ({@bruto}, {Orderinformatie.Order-id}) > 10000 then 

    crred 

 else 

    nocolor 

 

▫ sla het rapport op als oefening0204.rpt 

 

▪ Maak een nieuw rapport met de tabel klanten 

▫ waarbij de paginanummers beurtelings links en rechts staan 

▫ er zijn een aantal alternatieve oplossingen 

▫ één paginanummer paginabreed waarbij de uitlijning alternerend links of rechts is 

En als Formula: 

 

 if pagenumber mod 2 = 0 then 

    crLeftAligned 

 else 

    crRightAligned 

 

▫ twee pagina secties a en b met een pagenumber in de eerste links en de tweede rechts; 

deze onderdrukken we dan beurtelings 

▫ sla het rapport op als oefening0205.rpt 

 

▪ Maak een rapport met de tabel producten 

▫ Plaats een plaatje (bijvoorbeeld uit de Windows map) in de page header  

▫ Kies bij de Section Expert voor Underlay following sections 

▫ sla het rapport op als oefening0206.rpt 

14 Opslaan van een rapport in een ander formaat 

14.1 Inleiding 

Het is in Crystal Reports mogelijk een rapport op te slaan in verschillende formaten. We kiezen daarvoor 

File  Export  Export Report 

 

 

 

We krijgen dan het volgende scherm: 

 

 

 

En hier kunnen we te kust en te keur gaan. 

 

Bijzondere aandacht verdient hierbij: de optie Report Definition. Met deze optie is het mogelijk een 

gedetailleerde beschrijving van het rapport naar een tekstbestand te laten wegschrijven. Zie voorbeeld: 

 Crystal Report Professional v10.0 (32-bit) - Report Definition 

 

 1.0 File Information 

 

 Report File:  

 Version: 10.2 

 

 2.0 Record Sort Fields 

 

 3.0 Group Sort Fields 

 

 4.0 Formulas 

 

 4.1 Record Selection Formula 

 {Leveranciers.Land} = "Canada" 

 

 4.2 Group Selection Formula 

  

 

 4.3 Other Formulas 

 

 5.0 Sectional Information 

 

 5.1 Page Header Section 

 Visible, Keep Together 

 

 Bedrijf 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 

 

 Plaats 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 

 

 5.2 Page Footer Section 

 Visible, New Page After, Keep Together, Print At Bottom of Page 

 

 5.3 Report Header Section 

 Visible, New Page Before 

 

 5.4 Report Footer Section 

 Visible, New Page After 

 

 5.5 Group Header Section 

 Visible 

 

 GroupName ({Leveranciers.Land}) 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 

  

 5.6 Group Footer Section 

 Visible 

 

 5.7 Details Section 

 Visible 

 Subsection.1  

  Visible, Keep Together 

 

 {Leveranciers.Bedrijf} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 

 

 {Leveranciers.Plaats} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

14.2 Opgaven 

▪ Maak een rapport van de tabel Leveranciers  

▫ plaats de velden bedrijf en plaats in de detailsectie 

▫ sla dit op als een HTML bestand 

▫ bekijk het resultaat in de browser 

▫ sla het rapport op als oefening0301.rpt 

 

▪ Gebruik het vorige rapport  

▫ groepeer op Land 

▫ maak een selectie op Canada 

▫ maak een export van Report Definition naar een txt bestand op de desk top 

▫ bekijk dit tekstbestand 

▫ sla het rapport op als oefening0302.rpt 

▫ sla het tekstbestand op als oefening0302.txt 

15 Eindopdracht: maken van een complex rapport 

15.1 Inleiding 

Zoals eerder vermeld gaan we voor het maken van een rapport te werk volgens een vast stramien: 

 

▪ welke informatie wordt er gevraagd? 

▪ welke database(s) moeten we gebruiken? 

▪ welke tabellen/queries daaruit hebben we nodig? 

▪ welke relaties liggen er tussen tabellen? 

▪ welke velden selecteren we? 

▪ welke gegevens moeten we afleiden uit welke velden? 

▪ welke restricties passen we toe op welke velden of afleidingen? 

▪ hoe groeperen we? 

▪ Welke sorteervolgorde kiezen we? 

▪ welke cumulatieve bewerkingen (totalen, gemiddelden etc.) voor welke groepen van gegevens? 

▪ welke opmaak? 

15.2 Informatie-analyse 

De informatievraag 

Stel dat we een rapport willen dat per verkoper een overzicht van de jaarlijkse omzet per productgroep 

geeft. De jaarlijkse omzet willen we opgedeeld zien in kolommen voor de bruto omzet, de netto omzet, 

de verleende totaalkorting in procenten (lees: de korting per productgroep) en de gemiddelde levertijd 

per productgroep. Voor de bruto-omzet willen we zowel het totaal per verkoper als het subtotaal per 

jaar. 

Van elke verkoper willen we ook de persoonlijke en de bedrijfsgegevens vermeld zien. Ten slotte moet 

het rapport een grafiek bevatten waarin voor de specifieke verkoper de omzetjaren, productgroepen en 

bruto-omzet tegen elkaar worden afgezet. 

 

De database 

Als database gaan we gebruiken de standaard meegeleverde database: Noordenwind.mdb 

 

Tabellen  

Als we de bovenstaande informatievraag analyseren op benodigde gegevens, dan blijkt dat deze uit de 

volgende tabellen afkomstig zijn: Werknemers, Orders, Orderinformatie, Producten, Categorieën. 

 

Velden 

 

Werknemers: 
Achternaam +voornaam 
Foto 
geboortedatum 
in dienst 
toestelnummer 
adres 
plaats 
postcode 
telefoonnummer thuis 
provincie 

Orders: 
Orderdatum 
Leverdatum 

Orderinformatie: 
Prijs per eenheid 
hoeveelheid 

Producten: 
alleen voor de koppeling 

Categorieën: 
Categorienaam 

 

 

Relaties 

De genoemde tabellen zijn via onderstaande velden aan elkaar te relateren: 

▪ Werknemers.werknemer-id   <-> Orders.werknemer-id 

▪ Orders.order-id   <-> Orderinformatie.order-id 

▪ Orderinformatie.productnummer <-> Producten.productnummer 

▪ Producten.categorienummer <-> Categorie.categorienummer 

 

Afleidingen van velden 

Een aantal gegevens is niet rechtstreeks af te leiden uit de inhoud van een veld, maar moet soms uit 

meer een of meer velden worden samengesteld of berekend.  

 

▪ Naam   =  achternaam & ", " & voornaam 

▪ Bruto omzet  = orderinformatie.hoeveelheid*orderinformatie.prijs per eenheid 

▪ Netto omzet  = orderinformatie.hoeveelheid*orderinformatie.prijs per eenheid * (1-

orderinformatie.korting) 

▪ Korting   = 1-som netto omzet per categorie/som bruto omzet per categorie 

▪ Korting2  = 1-som netto omzet per jaar / som bruto omzet per jaar 

▪ Levertijd   = orders.leverdatum – orders.orderdatum 

 

Restricties 

Restricties zijn in het geval van onze informatievraag niet nodig. We zouden evenwel bijvoorbeeld een 

restrictie op de verkopers kunnen aanbrengen door voor één speciale naam te kiezen. 

 

Cumulatieve bewerkingen 

Een aantal gegevens berekenen we door velden of afleidingen van velden te totaliseren: 

  

▪ Totale bruto omzet alle verkopen 

▪ Totaal bruto en netto omzet per productgroep per jaar per verkoper 

▪ Gemiddelde korting per productgroep per jaar per verkoper 

▪ Gemiddelde levertijd per productgroep per jaar per persoon 

▪ Totaal bruto en netto omzet per jaar 

▪ Totaal bruto omzet per verkoper 

 

Volgorde 

Gesorteerd per verkoper per jaar per productgroep 

15.3 Uitwerking in Crystal Reports 

15.3.1 Nieuw rapport  

Uiteraard beginnen we met het maken van een nieuw rapport.  

We kiezen voor de optie As a blank report. De Report Wizard zou ons ook een heel eind op weg 

helpen, maar geeft onvoldoende inzicht in de verschillende stappen. En het gaat er uiteindelijk om zelf 

expert te worden! 

15.3.2 Koppeling naar de database Noordenwind.mdb 

Voor het maken van een koppeling met de database Noordenwind verwijzen we terug naar Dag I. 

15.3.3 Tabellen selecteren 

 

 

▪ kies door dubbelklikken de tabellen Werknemers, Orders, Orderinformatie, Producten en 

Categorieën en klik op Close 

15.3.4 Relaties 

 

 

▪ leg de relaties met de muis: 

Werknemers.werknemer-id   <-> Orders.werknemer-id 

Orders.order-id   <-> Orderinformatie.order-id 

Orderinformatie.productnummer <-> Producten.productnummer 

Producten.categorienummer <-> Categorie.categorienummer 

 

Standaard zal Crystal Reports tabellen automatisch proberen te linken op basis van velden met 

gelijkluidende namen. Dit kan uiteraard problemen op leveren.  

15.3.5 Formula Fields: berekende velden 

Je hebt onderstaand scherm in beeld. Is dit niet het geval dan kun je het oproepen via View → Field 

Explorer 

 

 

 

 

▪ klik op Formula Fields en vervolgens op de knop New (Ctrl N) 

▪ tik in: Naam, en klik op OK 

▪ in het onderste gedeelte typ je nu: 

 

 {werknemers.achternaam} & ", " & {werknemers.voornaam} 

 

▪ klik op Check (alt+c) links bovenin om de formule te controleren 

▪ klik vervolgens op Save and close (ctrl+s) om de formule op te slaan en af te sluiten 

 

 

Let op:  

Je kunt veldnamen ook selecteren door linksboven eerste de tabellen uit te laten klappen via de + en 

vervolgens dubbel te klikken op het veld 

 

 

▪ maak nu ook nog de volgende formula fields / rekenvelden: 

 

bruto omzet:  {orderinformatie.hoeveelheid}*{orderinformatie.prijs per eenheid} 

netto omzet:  {orderinformatie.hoeveelheid}*{orderinformatie.prijs per eenheid} *  

   (1-{orderinformatie.korting}) 

levertijd:   {orders.leverdatum} – {orders.orderdatum} 

 

Korting en Korting2 zullen we later als Formula field moeten toevoegen 

15.3.6 Selecties 

 

 

▪ gebruik de Select Expert om alleen de achternaam Davolio te selecteren; het tonen van het rapport 

in Preview wordt hierdoor aanmerkelijk sneller 

15.3.7 Groeperen van gegevens 

▪ zet de werkbalk Supplementary Tool aan, via de gele sleutel rechts bovenin (in de versie 9 en 10 

staat deze balk al aan en is de volgende knop al beschikbaar) 

▪ klik in deze balk op de knop Insert group  en voeg de groep werknemer-id aan het rapport 

toe 

▪ toon evenwel @naam 

▪ herhaal de stap groeperen voor Orderdatum en kies hierbij voor de optie For each year 

▪ voeg ten slotte de groep Categorienaam aan het rapport toe 

▪ voeg de gekozen velden uit de tabel Werknemers toe aan de koptekst van @naam (voor de indeling 

gebruik je het achteraan bijgevoegde rapport) 

 

 

Let op:  

De orderdatum had ook op een andere manier kunnen worden omgezet naar een jaar of bijvoorbeeld 

een kwartaal, namelijk via formula fields gebaseerd op functies. Een formula field met de volgende 

inhoud: 

 

 year({Orders.Orderdatum}) 

Geeft het jaar als resultaat maar wel in het formaat 1,996.00. 

 

Een formula field met de volgende inhoud: 

 

“Kwartaal “ + Totext(datepart(“q”,{Orders.Orderdatum}),0) 

 

zet elke datum om naar een kwartaal. 

 

15.3.8 Summaries 

 

 

▪ je plaatst in de Group Header van @naam via Insert Summary de Sum van @bruto omzet 

 

 

Let op:  

Het resultaat komt automatisch in de voettekst van @naam terecht; je kunt het simpelweg  terugslepen 

naar de koptekst 

 

 

▪ in de Group Header van de groep Orderdatum plaats je achtereenvolgend vier tekstvelden met de 

knop Insert Text Object (de kop met AB) met de teksten bruto omzet, netto omzet, korting en gem. 

levertijd 

▪ in de detail sectie plaatsje de rekenvelden @bruto omzet en @netto omzet 

▪ via Insert Summary voeg je de Sum van de rekenvelden @bruto omzet en @netto omzet toe aan de 

Group Header van Categorienaam 

▪ voor de derde kolom maak je een extra Formula Field Korting met de formule: 

 

 (1-Sum({@netto omzet},{categorienaam})/Sum({@bruto 

omzet},{categorienaam}))*100 

Let op:  

Typ de formule niet helemaal zelf! In de formula editor ontdek je bij de Report Fields de volgende: 

 

 

 

Als je daar het stuk:  Group #3: Categorieën.Categorienaam – A Sum of Netto Omzet aanklikt, krijg je 

automatisch het stuk: Sum({@netto omzet},{categorienaam})!!! 

 

Let op:  

In plaats van deze formule te gebruiken kunnen we het ook oplossen met behulp van Weighted average 

with. 

 

 

▪ dit veld voeg je toe aan de Group Header van Categorienaam 

▪ via insert Summary voeg je de AVG (gemiddelde) van het rekenveld @Levertijd toe aan dezelfde 

Group Header in de vijfde kolom 

▪ het resultaat kopieer je naar de footer van Group Header van de groep Orderdatum in de vijfde 

kolom 

▪ voor onderaan de vierde kolom in de Group Footer van de groep Orderdatum maak je een extra 

Formula Field Korting2 met de formule (ook deze formule moet je niet helemaal zelf gaan typen, 

maar klik nu aan: Group #2: Orders.Orderdatum– A Sum of Netto Omzet etc.):  

 

 (1-Sum ({@Netto omzet}, {Orders.Orderdatum}, "annually")/Sum ({@Bruto 

omzet}, {Orders.Orderdatum}, "annually"))*100 

 

▪ dit veld voeg je toe aan de Group Footer van de groep Orderdatum in de vierde kolom 

15.3.9 Grafiek 

▪ We gaan nu een grafiek toevoegen aan de Group Header van @Naam 

▪ We voegen een grafiek in via het menu Insert → Chart en kiezen als type 3D riser 

▪ We kiezen vervolgens het tabblad data → advanced  

▪ We krijgen dan onderstaand beeld: 

▪ We maken daarin een aantal keuzen: 

▫ Bij Place Chart kiezen we For each @Naam 

▫ Rechtsboven in het keuzemenu kiezen we On change of 

▫ Aan de bovenste rechthoek voegen we met de muis toe: 

Orders.orderdatum – A 

Categorieën.Categorienaam – A 

▫ Aan de onderste rechthoek voegen we toe: 

Sum of @Bruto omzet 

▪ Dan kiezen we het laatste tabblad: Text: 

▪ Daarin maken we bovenstaande wijzigingen 

 

Let op:  

Als de grafiek klaar is, kun je met de rechter muisknop via Chart Expert weer terugkomen in 

bovenstaande schermen 

 

15.3.10 Opmaak 

▪ We trekken een rechthoek met behulp van de knop insert box 

▪ Voor het opmaken van de rechthoek selecteren we deze en klikken we op de  rechter muisknop 

▪ We kiezen dan Format Box 

▪ Zet vervolgens de fill color op Silver en zet Drop Shadow aan 

▪ Getallen (Rechter muisknop: Format field  customize) 

▪ Percentages (procentknop) 

▪ Levertijd (Rechter muisknop: Format field  customize  currency symbol) 

▪ Paginanummering (Field explorer  special fields  Page N of M) 

▪ Groepen om en om kleuren; section expert  group header III  color  

▪ Klik op de aangewezen knop.  

▪ Type daar de volgende code: 

 

 if remainder(GroupNumber,2)=0 then  

    crSilver  

 else  

    crNoColor 

 

Houden we niet van typen, dan kan het ook bijna volledig menugestuurd: 

▪ Klik onder Operators op Control structures 

▪ Dubbelklik op if x then y else z 

▪ Plaats de cursor na de if  

▪ Klik onder functions op math 

▪ Dubbelklik op Remainder 

▪ Plaats de cursor achter het haakje ( 

▪ Sluit math weer met een enkele klik 

▪ Klik vervolgens onder functions op print state 

▪ Dubbelklik op GroupNumber 

▪ Zet achter de komma het getal 2 en achter het haakje ) = 0 

▪ Plaats de cursor achter then 

▪ Sluit print state weer met een enkele klik 

▪ Klik onder fucntions op Color constants 

▪ Dubbelklik op Silver 

▪ Plaats de cursor achter else 

▪ Dubbelklik op NoColor 

 

Het resultaat zou er dan zo uit moeten zien: 

 

 if Remainder (GroupNumber,2 )=0 then  

  crSilver  

 else  

  crNoColor 

Hadden we hetzelfde effect willen bereiken in een detailsectie, dan moet we een kleine verandering 

aanbrengen: 

 

 if remainder(RecordNumber,2)=0 then  

    crSilver  

 else  

    crNoColor 

 

15.4 Extra opgaven 

▪ Geef een overzicht per leverancier van de betrokken producten. Het overzicht moet - verdeeld per 

jaar per categorie per product - de omzet geven er van uitgaande dat de leverancier de helft van de 

bruto-omzet als prijs heeft gekregen. 

 

▪ Doe hetzelfde voor de verzenders er van uitgaande dat een verzender 1 procent van de bruto-

omzet vraagt 

15.5 Relatieschema 

16 Voorbeeldrapport 

 

 

Davolio, Nancy 

Geboren: 

Toestelnr.: 

In dienst: 

Privé adres: 

Total bruto omzet  202.143,71 

12/8/68 

5/1/92 

5467 

507 - 20th Ave. E. 

Seattle 

98122 

(206) 555-9857 

WA 

Bruto omzet: Netto omzet: Korting: Gem. levertijd: 1996 

 9.632,00  8.847,88  8% Dranken  7 dagen 

 2.135,00  1.782,50  17% Fruit  5 dagen 

 355,20  355,20  0% Graanproducten  8 dagen 

 2.762,00  2.472,93  10% Kruiden  8 dagen 

 4.538,30  4.167,24  8% Vis  9 dagen 

 1.234,70  1.046,06  15% Vlees  7 dagen 

 8.212,80  7.642,95  7% Zoetwaren  8 dagen 

 9.919,00  9.449,76  5% Zuivel  7 dagen 

 38.789,00  35.764,52  8%  7 dagen 

Bruto omzet: Netto omzet: Korting: Gem. levertijd: 1997 

 13.644,00  13.112,08  4% Dranken  9 dagen 

 14.043,00  13.672,00  3% Fruit  8 dagen 

 6.639,75  6.183,08  7% Graanproducten  5 dagen 

 6.947,35  6.797,84  2% Kruiden  10 dagen 

 12.909,60  12.534,48  3% Vis  8 dagen 

 8.517,30  8.012,95  6% Vlees  8 dagen 

 14.998,88  14.297,47  5% Zoetwaren  7 dagen 

 19.833,70  18.538,20  7% Zuivel  7 dagen 

 97.533,58  93.148,08  4%  8 dagen 

Bruto omzet: Netto omzet: Korting: Gem. levertijd: 1998 

 25.556,00  24.639,40  4% Dranken  7 dagen 

 4.276,45  4.251,75  1% Fruit  8 dagen 

 2.179,00  1.927,63  12% Graanproducten  8 dagen 

 4.324,50  4.290,80  1% Kruiden  7 dagen 

 7.499,15  7.442,44  1% Vis  8 dagen 

 6.594,07  5.979,47  9% Vlees  4 dagen 

 7.131,86  6.628,50  7% Zoetwaren  7 dagen 

 8.260,10  8.035,02  3% Zuivel  8 dagen 

 65.821,13  63.195,01  4%  7 dagen 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 102 

17 Geavanceerd formule gebruik 

In de basiscursus is al uitvoerig met simpele formules gewerkt. In dit hoofdstuk willen we een aantal 

voorbeelden geven van geavanceerder gebruik. We zullen trouwens bij deze voorbeelden ontdekken dat 

de moeilijke constructies vaak ook te vervangen zijn door simpeler varianten met behulp van een 

bestaande functie. 

17.1 Bijhouden hoogste orderregelsbedrag, datum en klant: if then 

Voor het navolgende voorbeeld gebruiken we de tabellen Orderinformatie, Order en Klanten uit de 

database noordenwind. We willen in het rapport zien welke klant op welke datum de orderregel met het 

hoogste bedrag heeft. 

 

▪ We maken een Formula Field met de volgende inhoud: 

 

 currencyvar BonusAmount; // zorgt er voor dat BonusAmount afgebeeld 

wordt als bedrag!!! 

 stringvar HighestCustname; 

 Datetimevar DateBonusReached; 

 

 if {@bruto omzet}> BonusAmount then 

    ( 

    HighestCustname := {Klanten.Bedrijf}; 

    DateBonusReached := {Orders.Orderdatum}; 

    BonusAmount := {@bruto omzet} 

    ); 

 //let op de haakjes!!!!! 

 "Het hoogste bedrag is  " & totext(BonusAmount) & " door " & 

HighestCustname & "op " & 

 totext(DateBonusReached,"d/M/yyyy") 

 

▪ Dit veld plaatsen we in de Report footer 

▪ De overige secties verbergen we.  

 

Het eindresultaat moet er dan als volgt uitzien: 

 

Het antwoord klopt trouwens niet. Er zijn twee klanten met een grootste orderregel van 15.810. We 

kunnen dit antwoord tevoorschijn krijgen met behulp van de functie Nthlargest (categorie Summary). De 

grammatica van deze functie is als volgt: NthLargest (10,{@bruto} ). Dit geeft aan welke bruto bedrag de 

tiende grootste is. 

17.2 Streepjes uit telefoonnummer halen: While Do loop 

In het volgende voorbeeld gebruiken we een While Do loop om net zolang van links naar rechts door 

een telefoonnummer te lopen totdat alle streepjes er uit zijn. We hebben de tabel Customers uit de 

database Xtreme2000 gebruikt.  

 

In het veld Phone staan meerdere streepjes. Het Formula Field om die streepjes eruit te halen ziet er zo 

uit: 

 StringVar NewPhone := {Customer.Phone}; 

 

 While Instr(NewPhone,"-")>0 Do 

 ( 

    NewPhone := Left(NewPhone, Instr(NewPhone,"-")-1) & 

    Right(NewPhone, Length(NewPhone) - Instr(NewPhone,"-")); 

 ); 

 NewPhone 

 

Het rapport ziet er zo uit: 

 

 

Mooi voorbeeld van een While Do loop, maar hetzelfde hadden we ook kunnen doen met de replace 

functie.  

17.3 Formulas om werkdagen tussen twee data te berekenen 

Voor het onderstaande voorbeeld hebben we de tabel Orders uit Noordenwind gebruikt. We willen het 

verschil in werkdagen berekenen tussen de Orderdatum en de Leverdatum. De oplossing bestaat uit 

twee delen. In het eerste deel nemen we de eventuele feestdagen. 

 

 //Vakantiedagen, moet in de report header van het rapport 

 DateVar Array Holidays := [ 

 Date (1996,07,10),  

 Date (1997,12,25), 

 Date (1997,12,31)  

 ]; 

 0 

 
In het tweede deel berekenen we hoeveel werkdagen er tussen twee datums zitten waarbij we er eerst 

de zater- en zondagen vanaf trekken. Vervolgens kijken we of er van de de tussenliggende dagen ook 

dagen bij de feestdagen horen. Die trekken we vervolgens ook nog af van het totaal. 

 //Hoofd formula, in de detailsectie 

 Local DateVar Start := datevalue({Orders.Orderdatum});  

 Local DateVar End := datevalue({Orders.Leverdatum});  

 Local Numbervar Hol; 

 DateVar Array Holidays; 

 

 numbervar diff; 

 diff:=datediff("d",Start,End)- 

 DateDiff ("ww", Start, End, 7) – 

 DateDiff ("ww", Start, End, 1); 

 

 Local NumberVar i; 

 For i := 1 to Count (Holidays) 

 do (if DayOfWeek ( Holidays[i] ) in 2 to 6 and  

 Holidays[i] in start to end then Hol:=Hol+1 ); 

 diff - Hol 

17.4 Opgaven 

▪ Maak een nieuw rapport met de tabellen orders en orderinformatie uit de database noordenwind 

▫ Maak een formula field bruto dat de velden [prijs per eenheid] en [hoeveelheid] met elkaar 

vermenigvuldigt. 

▫ Maak een formula field top10 met als inhoud: NthLargest (10,{@bruto}) 

▫ Plaats beide velden in de detailsectie 

▫ Verberg de detailsectie via de section expert met als voorwaarde {@bruto}<{@top10} 

▫ sla het rapport op als oefening0601.rpt 

 

▪ Maak een nieuwe rapport met de tabel leveranciers uit de database Noordenwind 

▫ Probeer aan de hand van het bovenstaande voorbeeld een formula field te maken dat alle 

spaties uit het veld [bedrijf] te halen.  

▫ sla het rapport op als oefening0602.rpt 

▫ probeer het probleem ook op te lossen met een formula field dat de Replace functie 

gebruikt 

▫ gebruik de helpfunctie om uit te vinden hoe je dat met deze functie moet aanpakken 

▫ sla het rapport op als oefening0603.rpt 

18 Parameters 

18.1 Parameters maken 

Parameters gebruiken we in Crystal Reports om de gebruiker zelf input te laten geven. Een parameter 

doet op zichzelf niets. Het gebruik er van wordt pas zinvol als we de parameter bijvoorbeeld koppelen 

aan een voorwaarde via de Select Expert: 

 

 

 

Hierin staat {?land} voor de parameter. 

 

Ook kunnen we parameters gebruiken om de opmaak te beïnvloeden. Bijvoorbeeld met de Formula:  

 

 {?Details}="nee"  

 

(waarbij de parameter de opties ja en nee kent). 

We maken parameters via de Field Explorer, met een klik op de rechter muisknop op Parameter fields of 

door een klik op de knop New. Het volgende scherm verschijnt dan: 

 

 

 

Bovenin geven we een naam, kiezen we het type en bepalen we of de waarden wel of niet uit een tabel 

moeten komen (Static of Dynamic) 

Let op:   

Vanaf versie 11 kunnen deze dan ook dynamisch bijgewerkt worden als we een verandering in de tabel 

aanbrengen; in oudere versies gebeurt dit dus niet! 

 

Bij Value options kunnen we ook allerlei zaken instellen: bijvoorbeeld Default values, allow multiple 

values, optional prompt. 

 

Als we nu een gemaakte parameter koppelen aan bijvoorbeeld een filter verschijnt het volgende scherm 

als we de Print preview oproepen: 

 

 

 

In dat scherm kunnen we dan onze keuzes maken. Willen we dat scherm later opnieuw oproepen, dan 

drukken we op F5 of klikken op de knop Refresh. 

 

Let op: 

Bij een dynamische parameter worden standaard alleen de eerste 500 unieke waarden getoond. Dit 

kunnen we overigens aanpassen via de Registr. Voor Windows 32 bits ziet dat er zo uit: 

 [HKEY_LOCAL_MACHINE\SOFTWARE\Business Objects\Suite 12.0\Crystal 

Reports\DatabaseOptions\LOV]  

"MaxRowsetRecords"="5000"  

 
Voor Windows 64 bits zo: 
 
 [HKEY_LOCAL_MACHINE\Software\Wow6432Node\Business Objects\Suite 

12.0\Crystal Reports\DatabaseOptions\LOV]  

 "MaxRowsetRecords"="5000"  

18.2 Parameter opnemen in een filter 

Als we een parameter opnemen in een filter, moeten we eraan denken dat de gebruiker ook niets kan 

kiezen. In dat geval moeten we in het filter de formule aanpassen. Als de parameter van het type string is 

doen we dat als volgt: 

 

 {klant.land} = {?land} or {?land} = "" 

 

In andere gevallen gebruiken we: 

 

 {klant.land} = {?land} or not hasvalue({?land}) 

 

Als een parameter opgenomen is in een filter, zien we dit ook terug in de Field Explorer. De bovenste 

parameter is opgenomen in een filter. 

 

 

18.3 Parameters invullen ook anders 

Vanaf versie 2008 kan dat ook door aan de linker kant het blokje parameters te kiezen en deze 

parameters daar dan in te vullen: 

18.4 Gekozen parameters tonen 

Als een gebruiker de parameters eenmaal ingevuld heeft, is het goed dat hij de gekozen waarden 

terugziet in bijvoorbeeld de Report Header. Bij simpele parameters kunnen we dat oplossen door deze 

gewoon daar te plaatsen, met een labeltje ervoor (zie pijl). 

 

 

 

Bij complexere parameters waarbij bijvoorbeeld voor meerdere waarden gekozen kan worden, wordt 

het een ander verhaal. 

18.5 Meerdere parameterwaarden tonen: For to loop  

In het volgende voorbeeld gebruiken we de tabellen Produkten, orderinformatie, orders uit de database 

noordenwind. We maken een rapport dat per product de totale omzet toont, met daarbij een parameter 

waarmee we meerdere producten kunnen selecteren. Vervolgens willen we de gemaakte keuzen voor de 

parameter in de Report Header laten afdrukken. De gekozen mogelijkheden uit de parameter worden 

opgeslagen in een Array. Deze Array kunnen we uitlezen met een loop functie: for i:=1 to 10 

 

Dit vereist een Formula Field met de volgende inhoud: 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 110 

 numbervar teller; 

 stringvar boodschap; 

 boodschap:= "U hebt geselecteerd: "; 

 for teller := 1 to count({?Product}) step 1 do // de for loop leest de 

Array uit 

 // count({?Product}) telt het aantal elementen uit de Array 

 ( 

    boodschap:= boodschap & {?Product}[teller] + ", " 

 ); 

 left(boodschap, length(boodschap)-2) // left functie verwijdert de 

laatste komma en spatie 

 

Voorbeeld: 

 

 

 

Bij een meervoudige numerieke parameter wordt de formule anders.  

 

 WhilePrintingRecords; 

 Local NumberVar Array InputNum := {?AnyNumericPrompt}; 

 //The last field above is the one you replace with your prompt field 

  
 Local StringVar str := ""; 

 Local NumberVar strLen := Count (InputNum); 

 Local NumberVar i; 

        For i := 1 to strLen 

        Step + 1 

        Do (str := str +', '+ ToText(InputNum [i],0,'')); 

  
 str [3 to Length(str)] 

 

 

18.6 Parameter met meerdere keuzes sorteren 

Als men meerdere waarden kiest, maar dat niet in alfabetische volgorde doet, moet de array eigenlijk 

gesorteerd worden. En dat kan met één van de volgende Formula fields: 

 stringvar array sort; 

 numbervar loop1; 

 numbervar loop2; 

 stringvar temp; 

 sort:= {?product}; 

 for loop2:=1 to ubound(sort)-1 do( 

    for loop1:=1 to ubound(sort)-loop2 do( 

 

        if sort[loop1] > sort[loop1+1] then 

            (temp := sort[loop1]; 

            sort[loop1] := sort[loop1+1]; 

            sort[loop1+1] := temp))); 

 Join(sort,", ") 

 

Of: 

 

 //voor het sorteren van de parameter 

 

 Local stringvar array myList:={? product};  

 Local stringvar temp; 

 Local numbervar i; 

 Local numbervar j; 

 for i:= 1 to count(myList)-1 do  

    for j:= 1 to count(MyList)-i do  

    if MyList[j] > MyList [j+1] then (  

        temp := MyList[j]; 

        MyList [j] := MyList [j + 1]; 

        MyList [j+1] := temp;); 

 join(mylist,", ") 

18.7 Paramater met ranges 

Je kunt bij Parameters ook kiezen voor Range. Je moet dan twee dingen invullen: een onder- en een 

bovengrens. Bijvoorbeeld 10 en 20. Via het filter bepaal je dan of de geselecteerde waarden binnen of 

buiten de range moeten vallen. Via Multiple values is het zelfs mogelijk meerdere ranges te kiezen.  

Om een range zichtbaar te maken moeten we gebruik maken van een dergelijke formule: 

 

 //voor het tonen van een Range 

 “Van “ & minimum(range) & “ tot “ & maximum(range) 

 

We kunnen bij een parameter ook aangeven dat we meer ranges willen kunnen instellen. De formule 

wordt dan een stuk ingewikkelder: 

 //voor het tonen van meer Ranges met parameter test 

 WhileReadingRecords; 

 NumberVar NumRanges := ubound( {?test} ); 

 NumberVar i := 0; 

 StringVar outStr := ""; 

 for i := 1 to  NumRanges  

  step 1  

  do 

    ( 

      stringVar maxCN := totext(Maximum({?test}[i]),0); 

      stringVar minCN := totext(Minimum({?test}[i]),0); 

      outStr := outStr & iif(outStr <> "", chrw(13), "") & 

      iif(len(minCN)>0, minCN, "The beginning") & 

      ' to ' &  

      iif(len(maxCN)> 0, maxCN, "The End"); 

    ); 

 outStr; 

 

Of een variant daarop: 

 

 BeforeReadingRecords; 

stringvar PrintYourPrompt; 

numbervar i; 

 

PrintYourPrompt := ""; 

for i := 1 to count( {?test} )  

step 1  
do 

   (PrintYourPrompt := PrintYourPrompt +  
 

   iif(i = 1,'',', ') +  

    
   iif(minimum( {?test} [i] ) = maximum( {?test} [i] ) , 

     totext( minimum( {?test} [i])),  

     totext ( minimum ( {?test} [i] ) ) + 

    
   ' - ' + totext ( maximum( {?test} [i] )))); 
 

if Length ( PrintYourPrompt ) > 0 then     

   PrintYourPrompt [ 1 to Length ( PrintYourPrompt ) ] 

18.8 Cascading parameters (versie XI en hoger) 

Vanaf versie XI is het mogelijk cascading parameters te maken waarbij de keuzemogelijkheden van de 

tweede afhangen van de eerste. Kijk naar het volgende voorbeeld met de tabel Klanten. In deze tabel 

zijn land en plaats hierarchisch van elkaar afhankelijk. Kiezen we voor Duitsland, dan zouden we 

vervolgens niet meer voor Parijs mogen kiezen.  

 

We lossen dat op door voor een dynamische parameter te kiezen. Bij Value kiezen we dan 

achtereenvolgend de velden Land en Plaats. Ten slotte klikken we achteraan onder Parameters om 

aparte parameters te maken. 

Met als resultaat: 

 

 

 

Ook als we alleen de plaats zouden gebruiken in het rapport, komt toch eerst de vraag naar land 

tevoorschijn. 

18.9 Opgaven 

▪ We gaan een parameter gebruiken om de omzet van bepaalde producten te selecteren 

▫ Gebruik de tabellen producten, orders en orderinformatie uit de database Noordenwind 

▫ Maak een formula field Bruto met als inhoud het product van de velden [prijs per eenheid] 

en [hoeveelheid] 

▫ Maak een dynamic parameter product die meerdere waarden (Allow multiple values = 

true) toestaat  

▫ met als inhoud de productnamen uit de tabel producten 

 

 

 

o Koppel via de Select Expert het veld productnaam uit de tabel producten aan de 

parameter product 

▫ Probeer het een en ander uit in de Print preview 

▫ Druk ook op F5 voor het verversen 

▫ sla het rapport op als oefening0701.rpt 

 

▪ Maak een rapport met de tabel klanten 

▫ Maak een dynamic parameter Land waarbij meerdere keuzen gemaakt kunnen worden. 

▫ Maak een formuleveld dat de gemaakte keuzen weergeeft  

▫ plaats deze in de Report Header 

▫ Probeer het een en ander uit in de Print preview 

▫ Druk ook op F5 voor het verversen 

▫ sla het rapport op als oefening0702.rpt 

 

▪ We kunnen ook parameterwaarden tonen met de Join functie.  

▫ Zoek zelf uit hoe met behulp van de helpfunctie. 

▫ sla het rapport op als oefening0703.rpt 

 

▪ We kunnen een parameter ook gebruiken om selectief te groeperen. 

▫ Maak een rapport met de tabellen Orders, Orderinformatie en Klanten. 

▫ Maak een formula field Bruto met als inhoud het product van de velden [prijs per eenheid] 

en [hoeveelheid]. 

▫ Maak een parameter die laat kiezen voor groeperen op Ordernummer of Klant. 

▫ Maak een Formula Field dat met behulp van een if then constructie de input uit de 

parameter omzet naar het veld {klanten.bedrijf} resp. {orders.order-id}. 

▫ Groepeer op dit formula field. 

▫ Maak groepstotalen op basis van het formula field Bruto. 

▫ Test de werking door bij de parameter input te kiezen voor resp. ordernummer en klant. 

▫ Sla het rapport op als oefening0704.rpt. 

 

▪ Maak een rapport waarbij de Detail Section selectief onderdrukt kan woorden door een Parameter 

▫ Sla het rapport op als oefening0705.rpt.  

 

▪ Maak een rapport op basis van de tabellen Categorieen en Producten 

▫ Maak een cascading parameter waarbij we achtereenvolgend categorienaam en producten 

kunnen kiezen (bij versie 2008 kan zich hier een probleem voordoen: in een bepaalde 

variant van versie 2008 werkt de cascading parameter alleen als beide velden uit dezelfde 

tabel komen; dit is op te lossen door met een SQL Command te werken waarbij beide 

velden in dezelfde view geplaatst worden). 

▫ Sla het rapport op als oefening0706.rpt. 

19 Running totals 

Een Running Total wordt in het Nederlands ook wel een doorlopend totaal of een cumulatief genoemd. 

Voorbeeld:  het cumulatieve bedrag van twaalf maanden loon. 

 

We kunnen een Running Total maken door met de rechter muisknop op een veld in Detailsectie te 

klikken. Ook kunnen we in de Field Explorer een nieuw Running Total field maken door rechts te 

klikken op Running Total Fields. 

 

Een Running Total veld kunnen we in het rapport invoegen zoals een databaseveld. Zonder formules of 

variabelen kunnen we een waarde laten accumuleren met het voortschrijden van het rapport. 

 

 

 

In het onderstaande voorbeeld maken we op basis van de tabel Orderinformatie uit de database 

Noordenwind een rapport dat naast het ordernummer en de bedragen ook Running Totals geeft die resp. 

alles optellen, alleen de bedragen boven de 1000 en allebei steeds opnieuw beginnen voor elk nieuw 

ordernummer. 

 

▪ We groeperen eerst op ordernummer  

▪ maken een formula field Omzet dat Prijs per eenheid vermenigvuldigt met Hoeveelheid. 

 

 

 

Wanneer wordt 

het Running 

Total opgehoogd 

Wanneer wordt 

het Running 

Total gereset 

Het eerste Running Total field stellen we als volgt in: 

 

 

 

▪ Het tweede Running Total als volgt: 

▪ Als Formula gebruiken we dan:  

 

 {@omzet}>1000 

 

Beide Running Total velden plaatsen we in het rapport. Het eindresultaat ziet er dan als volgt uit. 

 

 

19.1 Opgaven 

▪ Maak een rapport met de tabellen orders en orderinformatie 

▫ Maak een groepering op orderdatum per kwartaal 

▫ Maak een formula field Omzet dat de omzet berekend: hoeveelheid * prijs 

▫ Plaats dit in de detailsectie 

▫ Maak een Running Total op basis van het formula field Omzet 

▫ Plaats dit in de detailsectie 

▫ Pas het Running Total zo aan dat er een reset plaatsvindt per groep  

▫ en alleen de omzetbedragen boven de 1000 worden meegeteld 

▫ sla het rapport op als oefening0801.rpt 

 

 

 

▪ Maak het bovenstaande voorbeeld na: een rapport met de tabellen Orders en Orderinformatie.  

▫ Maak een Formula field bruto dat de omzet berekent (hoeveelheid * prijs) 

▫ Plaats dit in de detailsectie 

▫ Groepeer op order-id 

▫ Zet het groepslabel in de footer van de groep 

▫ Sommer op bruto  

▫ Plaats het totaal in de footer van de groep 

▫ Verberg group header en detailssectie 

▫ Maak een Running total dat alleen de bruto bedragen boven de 500 meetelt en per groep 

een reset krijgt 

▫ Plaats dit Running total in de group footer 

▫ Maak een Formula field dat het Running total deelt op het groepstotaal en vermenigvuldig 

dit met 100 om er een percentage van te maken 

▫ Plaats dit Formula field ook in de group footer 

▫ Maak een numeric parameter Omzet 

▫ Vervang de 500 in het Running total door de parameter zodat de gebruiker zelf kan kiezen 

▫ Probeer het een en ander uit in de Print preview 

▫ sla het rapport op als oefening0802.rpt 

 

▪ Maak een rapport met de tabel Klanten 

▫ Plaats het veld Bedrijf in de detailsectie 

▫ Maak een Running total teller zoals in onderstaan scherm 

▫ Zorg er voor dat de detailsectie om en om gekleurd wordt op basis van deze teller 

 

 if {#Teller} mod 2 = 0 then 

    crgray 

 else 

    nocolor 

 

▫ sla het rapport op als oefening0803.rpt 

 

▪ Maak een rapport met de tabellen Producten.  

▫ Plaats Prijs per eenheid in de detailsectie. 

▫ Maak een Running total op basis van dit veld. 

▫ Kies hier voor MAXIMUM 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 121 

 

 

▫ Pas de opmaak aan van het eerste veld. 

 

 

 

▫ Maak bij Background de volgende Formula: 

 if {Producten.Prijs per eenheid}>={#RTotal0} then 

    crred 

 else 

    nocolor 

 

▫ Bekijk het resultaat in de Print preview. 

 

Als resultaat zien we het voortschrijdende maximum rood gekleurd. 

 

▫ Sla het rapport op als oefening0804.rpt. 

20 Kruistabellen 

20.1 Inleiding 

In een kruistabel kruisen we simpel gezegd twee – of meer - kolommen met elkaar en op de kruispunten 

plaatsen we een derde veld – kunnen er ook meer zijn - dat daar dan geaccumuleerd wordt als sum, 

count average etc. 

 

In het volgende voorbeeld gebruiken we de tabellen Orderinformatie, Orders, Categorieën en 

Producten uit de database Noordenwind. We willen een kruistabel waarbij de gesommeerde omzet 

wordt afgezet tegen de categorienaam (kolom) en het kwartaal (orderdatum als rij). 

 

▪ Maak een rapport met de tabellen Orderinformatie, Orders, Categorieën en Producten. 

▪ We hebben weer het Formula field omzet (hoeveelheid * prijs) nodig. 

▪ Klik op de knop insert cross-tab (een kruistabel kan alleen maar in secties staan waar ook 

summaries kunnen staan: Report header en footer en Group header en footers). 

▪ Rechtsklik op de lege cross-tab. 

▪ Kies Cross-tab expert. 

 

 

 

▪ Maak bovenstaande keuzes. 

▪ Zet de group options voor Orderdatum op For each month. 

▪ Kies het tabblad Customize Style. 

▪ Zet Row totals on left en Columns totals on top uit. 

▪ Klik op OK. 

▪ Ga naar Print preview. 

▪ Kies via Page setup voor Landscape. 

▪ Zet van alle velden de lettergrootte op 8. 

 

En dan krijgen we dit voorbeeld: 

▪ Als we een groep maken op Categorienaam en we plaatsen de kruistabel eveneens daarin, dan 

krijgen we het tweede plaatje: 

 

 

 

20.2 Opgaven 

• Maak een kruistabel met de tabellen klanten, orders en orderinformatie uit de database 

Noordenwind 

▫ We maken weer een Formula field bruto (hoeveelheid * prijs) 

▫ In een kruistabel zetten we klanten (verticaal) en jaren (horizontaal, op basis van 

orderdatum) tegen elkaar af.  

▫ Op de knooppunten laten we het totaal van de bruto omzet berekenen. 

▫ sla het rapport op als oefening0901.rpt 

 

▪ We maken nu een rapport met de tabellen Orderinformatie, Order, Categorieën en Producten 

▫ We maken nu een kruistabel waarbij de som van de omzet wordt afgezet tegen de 

categorienaam (kolomkop) en de maand (orderdatum als rijkop, met group options op For 

each month)  

▫ We willen nu dezelfde maanden uit de verschillende jaren bij elkaar hebben 

▫ We hebben twee extra formula fields nodig 

▫ ééntje die de maand en dan het jaar (met month en year) uit de datum haalt  

 

 totext(month({Orders.Orderdatum}),"00") & " " & 

totext(year({Orders.Orderdatum}),"0000") 

 

▫ en een tweede die het precies andersom doet 

 

 totext(year({Orders.Orderdatum}),"0000") & " " & 

totext(month({Orders.Orderdatum}),"00") 

 

▫ we groeperen dan op de eerste  

▫ en tonen de tweede 

 

 

 

Let op:   

We kunnen maandnummers desgewenst van een voorloopnul voorzien met behulp van een formule:  

 

 right("00" +totext(month({Orders.Orderdatum}),0),2),  

 

of nog korter:  

 

 totext(month({Orders.Orderdatum}),"00") 

 

▪ We bekijken de uitkomst via Print preview: 

▪  sla het rapport op als oefening0902.rpt 

 

Voor mensen met tijd te veel: we kunnen ook een kruistabel maken op basis van formula fields. Dit is in 

het onderstaande voorbeeld gedaan. Er zijn twee parameters gebruikt. De eerste om te bepalen over 

welke jaren de kruistabel gemaakt moet worden en de tweede om te bepalen of de Group Section voor 

de klant wel of niet getoond moet worden. Er wordt getoond de bruto omzet per product per klant. 

 

▪ Maak een rapport met de tabellen Orderinformatie, Order, Categorieën en Producten. 

▫ Maak een numeric parameter welkjaar. 

▫ Maar een Formula field welkjaar met de parameter als inhoud:  

 

 {?welkjaar} 

 

▫ Maak nog 4 Formula fields, resp. welkjaar-1 tot en met welkjaar-4 met als inhoud:  

 

 {?welkjaar}-1 etc. 

 

▫ Maak een Formula field gekozen jaar met als inhoud: 

 

 if year({Orders.Orderdatum})= {?welkjaar} then 

    {@bruto} 

 else 

    0 

 

▫ Maar nog 4 Formula fields, resp. gekozen-1 tot en met gekozen -4 met als inhoud: 

 

 if year({Orders.Orderdatum})= {?welkjaar}-1 then 

    {@bruto} 

 else 

    0 

 

▫ Maak drie groepen, resp. categorienaam, productnaam en bedrijf 

▫ Plaats de velden gekozenjaar etc in de detailsectie 

▫ Totaliseer deze velden 

▫ Plaats de uitkomst in alle group footers 

▫ Plaats de velden welkjaar etc in de group footer van groep 1 

▫ Verberg de detailsectie 

▫ Bekijk de uitkomst via Page preview 

▫ Kies als startjaar 1999 

▫ sla het rapport op als oefening0903.rpt 

21 Geavanceerde kruistabelfuncties 

We kunnen op verschillende manieren nog meer gegevens aan onze kruistabellen toevoegen. In dit 

hoofdstuk wordt een aantal geavanceerde kruistabelfuncties beschreven. 

21.1 Calculated Members 

We kunnen nieuwe rijen of kolommen aan de kruistabel toevoegen door een Calculated Member in te 

voegen. Deze rijen of kolommen kunnen aangepaste berekeningen bevatten (bijvoorbeeld het verschil in 

verkoop tussen twee regio's) of kunnen voor de opmaak worden gebruikt (bijvoorbeeld om de drie regels 

een lege rij invoegen om de leesbaarheid te vergroten). 

 

We hebben de keuze om geheel naar wens Calculated Formula's te maken en kunnen aangeven of deze 

eenmaal moeten worden weergegeven of moeten worden herhaald in de kruistabel. 

 

Verwante onderwerpen 

 

• Calculated Formulas 

• Header Formula's 

• Insertion Formula's 

• Column /Row Value Formula's 

21.1.1 Een Calculated Member toevoegen aan de kruistabel 

Voor onderstaande voorbeeld hebben we de tabellen Categorieën, Producten, Orders en 

Orderinformatie gebruikt.  

 

▪ Maak twee Formula Fields: 

 

 Omzet:{Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid} 

 

 Jaar: year({Orders.Orderdatum}) 

 

▪ Maak een kruistabel als volgt: 

 

 

 

▪ Klik op OK. 

▪ Klik in de kruistabel met de rechtermuisknop op de rij- of kolomkop die we willen gebruiken als 

eerste waarde voor het Calculated Member. 

 

 

 

Als we bijvoorbeeld een Calculated Member willen maken voor het verschil tussen 1996 en 

1997,gebruiken we de formule Verschil 1997 - 1996. In dit voorbeeld klikken we met de 

rechtermuisknop op de kop 1996. 

▪ Wijs in het snelmenu de optie Calculated Member aan.  

▪ Selecteer de gekozen waarde als de eerste waarde.  

Klik in de kruistabel met de rechtermuisknop op 1997 als tweede waarde. 

▪ Wijs in het snelmenu de optie Calculated Member aan. 

▪ Selecteer de gewenste berekening: Difference 1997 - 1996. 

 

In de kruistabel verschijnt een rij of kolom met daarin het resultaat van de geselecteerde berekening. 

 

▪ Klik met de rechter muisknop op één van de nieuwe velden: 

 

 

 

▪ Klik Edit Calculation Formula. 

 

We zien dan de formula: 

 

 //The calculation formula is used in place of the summaries in 

//Calculated Members. 

 // This calculation formula must return a/an Currency value. 

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1997.00), 

CurrentSummaryIndex) -  

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1996.00), 

CurrentSummaryIndex) 

We zien hier dan dat de functie GridValueAt het snijpunt kiest van de huidige rij (CurrentRowIndex) en 

de kolom waarin zich het jaar bevindt (GetColumnPathIndexOf(1996.00)) en van dat snijpunt de 

aanwezige Summary (CurrentSummaryIndex) kiest. Zo worden dus in deze formule voor elke regel de 

Summary 1997 en 1996 van elkaar afgetrokken. 

21.1.2 Een lege rij of kolom toevoegen aan de kruistabel 

▪ Klik in de kruistabel met de rechtermuisknop op de kop vóór de plek waar we de lege kolom willen 

invoegen. 

▪ Wijs in het snelmenu de optie Calculated Member aan. 

▪ Klik op Kolom invoegen. 

 

Er wordt een lege kolom in de kruistabel weergegeven. 

 

▪ Klik op één van de nieuwe belden. 

▪ Kies Calculated Member   Edit Calculation Formula. 

▪ Typ daar de Formula: 

 

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1998.00), 

CurrentSummaryIndex) -  

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1997.00), 

CurrentSummaryIndex) 

 

Het resultaat moet hetzelfde zijn als in de vorige paragraaf (21.1.1). 

21.1.3 De verwerkingsvolgorde van berekende leden wijzigen 

▪ Klik met de rechtermuisknop in de linkerbovenhoek van de kruistabel. 

▪ Wijs in het snelmenu de optie Geavanceerde berekeningen aan en klik op Calculated Member.  

 

De Expert voor Calculated Members  wordt weergegeven. 

 

 

 

Bij beide zien we nog geen naam.  

 

▪ Noem de eerste Kolom na 1996 en de tweede Kolom na 1997 

 

Bij Edit Insertion Formula zien we: 

 //The insertion formula determines where this calculated member is 

//inserted. It must return a Boolean value. 

 GetColumnGroupIndexOf(CurrentColumnIndex) = 1 and 

GridRowColumnValue("@year") = 1996.00 

 

Bij Edit Column Value Formula staat niets. Bij Edit Header Formula treffen we de eerder gemaakte 

headers aan. 

 

Via dit scherm hadden we ook een kolom kunnen toevoegen. Bijvoorbeeld na 1998. Daar hadden we 

dan de volgende Formula voor moeten gebruiken: 

 

 GetColumnGroupIndexOf(CurrentColumnIndex) = 1 and 

GridRowColumnValue("@year") = 1998.00 

 

▪ Gebruik de pijlen in het vak Calculated Members om de verwerkingsvolgorde van de Calculated 

Members te wijzigen. 

▪ Klik op OK om de wijzigingen op te slaan en terug te gaan naar het rapport.  

 

De waarden van de Calculated Members worden opnieuw berekend in de kruistabel. 

 

Opmerking: 

Wijzigingen in de volgorde waarin Calculated Members worden verwerkt, hebben geen invloed op het 

vernieuwen van de database. 

21.1.4 Calculating formula's 

Calculating formula's zijn wiskundige formules die door Crystal Reports worden gebruikt om de 

celwaarden van Calculated Members te bepalen. We kunnen een keuze maken uit de vooraf 

gedefinieerde formules of zelf formules ontwerpen in de Formula workshop. 

 

De vooraf gedefinieerde formules zijn als volgt: 

 

• Sum of 

• Difference of 

• Product of 

• Quotient of 

 

Deze zien er in de Formula workshop zo uit: 

 

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1998.00), 

CurrentSummaryIndex) +  

 GridValueAt(CurrentRowIndex, GetColumnPathIndexOf(1997.00), 

CurrentSummaryIndex) 

 

Uiteraard moeten we dan respectievelijk  -, * of / invullen. 

 

Crystal Reports bevat verder diverse functies die we kunnen gebruiken om zelf formules te ontwerpen. 

21.1.4.1 Calculation Formula bewerken 

▪ Klik in het Calculated Member dat we willen bewerken met de rechtermuisknop op een 

willekeurige cel, maar niet op de kop. 

▪ Wijs in het snelmenu de optie Calculated Member  aan. 

▪ Klik op Edit Calculation Formula.  

 

De Formuleworkshop wordt weergegeven. 

 

▪ Geef in de Formuleworkshop de gewenste berekeningsformule op. 

▪ Klik op Opslaan en sluiten om terug te keren naar het rapport.  

Het resultaat van de nieuwe berekeningsformule wordt in het Calculated Member weergegeven. 

21.1.5 Header Formula's 

Header Formula's bepalen de zichtbare rij- of kolomkoptekst van een Calculated Member. 

21.1.5.1 Edit Header Formula 

▪ Klik met de rechtermuisknop op de koptekst van het Calculated Member dat we willen bewerken. 

▪ Wijs in het snelmenu de optie Calculated Member aan. 

▪ Klik op Edit Header Formula.  

 

De Formuleworkshop wordt weergegeven. 

 

▪ Geef in de Formuleworkshop de gewenste naam op tussen aanhalingstekens. 

 

Opmerking: 

We kunnen de Header ook opgeven met een formule. Een Header Formula moet een tekenreeks 

retourneren. 

 

▪ Klik op Opslaan en sluiten om terug te keren naar het rapport.  

 

In de kop van het Calculated Member wordt het resultaat van de Header Formula  weergegeven. 

21.1.6 Waardeformules 

Als we in een formule naar een Calculated Member willen verwijzen, moeten we een waarde aan het 

lid toewijzen. Om deze reden kunnen we met een waardeformule waarden toewijzen aan Calculated 

Members. 

21.1.6.1 Edit Column / Row Value Formula 

▪ Klik met de rechtermuisknop op de Header van de Calculated Member die we willen bewerken. 

▪ Wijs in het snelmenu de optie Calculated Member aan. 

▪ Klik al naar gelang we een rij of kolom willen bewerken respectievelijk op Edit Row Value Formula 

of Edit Column Value Formula. 

 

De Formula workshop wordt weergegeven. 

 

▪ Geef in de Formula workshop de gewenste waarde op. 

 

Opmerking: 

We kunnen de waarde ook opgeven met een formule. Een waardeformule moet van hetzelfde type zijn 

als het groepeerveld voor rijen of kolommen waarin de formule wordt ingevoegd. 

 

▪ Klik op Opslaan en sluiten om terug te keren naar het rapport. 

 

In het Calculated Member wordt het resultaat van de Value Formula weergegeven. 

 

Opmerking: 

Als we de Header Formula van een Calculated Member niet bewerkt hebben, wordt het resultaat van de 

Value Formula in de zichtbare koptekst weergegeven. Wanneer we de Header Formula bewerkt 

hebben, wordt de Value Formula als de zichtbare koptekst hierdoor overschreven. 

21.1.7 Insertion Formula's 

Insertion Formula's  bepalen waar een Calculated Member in een kruistabel moet worden 

weergegeven. Meestal wordt een Calculated Member slechts eenmaal weergegeven. We kunt echter 

ook aangeven dat het lid op meerdere locaties of in een herhaald patroon moet worden weergegeven. 

21.1.7.1 Edit Insertion Formula 

▪ Klik met de rechtermuisknop op de koptekst van het Calculated Member dat we willen bewerken. 

▪ Wijs in het snelmenu de optie Calculated Member aan. 

▪ Klik op Edit Insertion Formula. 

 

De Formula Workshop wordt weergegeven. 

 

▪ Geef in de Formula Workshop de gewenste formule op. 

 

Met de volgende formule bijvoorbeeld, wordt een Calculated Member ingevoegd na het jaar 1998: 

 

 GridRowColumnValue("@year") = 1998.00 

 

▪ Klik op Opslaan en sluiten om terug te keren naar het rapport. 

 

Het Calculated Member wordt weergegeven op de locaties die in de Insertion Formula zijn 

gedefinieerd. 

21.2 Embedded Summary 

We kunnen een extra berekening aan de kruistabel toevoegen door een Embedded Summary in te 

voegen. In tegenstelling tot Calculated Members, worden er met Embedded Summaries geen rijen of 

kolommen aan de kruistabel toegevoegd. In plaats daarvan worden er extra berekeningen aan de cellen 

van de kruistabel toegevoegd. Zo kunnen we in een rapport met regionale verkoopcijfers bijvoorbeeld 

een berekening voor elke regio invoegen, waarin het percentage van de totale landelijke verkoop wordt 

weergegeven. 

21.2.1 Een Embedded Summary toevoegen aan de kruistabel 

▪ Klik in de kruistabel met de rechtermuisknop op een willekeurige cel, maar niet op de kop. 

▪ Wijs in het snelmenu de optie Embedded Summary aan. 

▪ Klik op Insert Embedded Summary. 

 

 

 

Een lege Embedded Summary wordt in de kruistabel weergegeven. 

 

▪ Klik in de Embedded Summary met de rechtermuisknop op de tekst Edit this Formula. 

▪ Wijs de optie Embedded Summary aan. 

▪ Klik op Edit Calculation Formula. 

 

De Formula Workshop wordt weergegeven. 

 

▪ Geef in de Formula Workshop de gewenste Calculated Formula op. 

 

 GridValueAt (CurrentRowIndex, CurrentColumnIndex ,0 ) / 

sum({@omzet})*100 

 

Deze Formula berekent de verhouding tussen elke lokale Summary en het generaal totaal. 

 

▪ Klik op Opslaan en sluiten om terug te keren naar het rapport. 

 

In de Embedded Summary wordt het resultaat van de nieuwe berekeningsformule weergegeven. 

21.2.2 De verwerkingsvolgorde van Embedded Summaries wijzigen 

Als de kruistabel meerdere Embedded Summaries bevat, kan de volgorde waarin deze worden 

verwerkt invloed hebben op het resultaat. We kunnen de volgorde waarin Embedded Summaries 

worden verwerkt, wijzigen in het dialoogvenster Embedded Summaries. 

 

▪ Klik met de rechtermuisknop in de linkerbovenhoek van de kruistabel. 

▪ Wijs in het snelmenu de optie Geavanceerde berekeningen aan.  

▪ Klik op Embedded Summary. 

 

Het dialoogvenster Embedded Summaries  wordt weergegeven. 

 

 

 

▪ Gebruik de pijlen in het vak Summaries om de verwerkingsvolgorde van de Summaries te wijzigen. 

▪ Klik op OK om de wijzigingen op te slaan en terug te gaan naar het rapport. 

 

De waarden van de Embedded Summaries  worden in de kruistabel opnieuw berekend in de nieuwe 

verwerkingsvolgorde. 

 

Opmerking: 

Wijzigingen in de verwerkingsvolgorde van Embedded Summaries hebben niet tot gevolg dat er een 

vernieuwing via de database wordt uitgevoerd. 

22 Optimaliseren van rapporten: performance 

22.1 Inleiding 

Met name bij grote database kan de performance van rapporten probelematisch worden. Vooral ook 

tijdens de ontwikkelfase is het bijna ondoenlijk om dan direct op de productiedatabase te werken. 

 

Hier zullen we ons verder beperken tot het geven van algemene regels. Als een algemene regel geldt: 

laat de database server zoveel mogelijk het werk doen.  

 

Het volgende voorbeeld hebben we gedaan met de tabel orderinformatie uit de database Noordenwind.  

 

▪ Uit de tabel plaatsen we de velden order-id, prijs per eenheid en hoeveelheid in de detailsectie. 

▪ Om te zien hoeveel van de query geëvalueerd wordt door de database server gaan we naar het 

menu Database  

▪ We kiezen we de optie Show SQL Query:  

 

 

 

22.2 SQL Expression fields 

Vuistregels  

 

▪ Vermijd selecties op basis van Formula Fields; Crystal Reports kan deze Formulas vaak niet 

omzetten naar de SQL query 

▪ Gebruik zoveel mogelijk SQL Expression velden 

▪ Vermijd het gebruik van Built-in functies zoals ToText in selectiecriteria; probeer deze te vervangen 

door een SQL functie 

▪ Vermijd het gebruik van string subscript functies zoals {klanten.bedrijf}[1] = “A”; zoek een 

vergelijkbare functie in SQL 

 

We gaan nu het een en ander toepassen. 

 

▪ We maken nu twee velden 

▪ Een formula field omzet dat hoeveelheid * prijs berekent 

▪ Daar maken we een filter op met > 1000 

▪ In de SQL query zien we dan: 

 

 

 

▪ Nu maken we een SQL expression field met dezelfde berekening; technisch ziet dat er ook iets 

anders uit 

 

 `Orderinformatie`.`Prijs per eenheid`*`Orderinformatie`.`Hoeveelheid` 

 

▪ Dezelfde selectie op het tweede veld geeft nu een heel ander beeld: 

Het criterium is opgenomen in het SQL statement en dat betekent dat de database server veel minder 

records hoeft te sturen. 

22.3 Server-based grouping 

Als je in het menu Database de optie Perform Grouping on Server aanvinkt, kan dat betekenen dat het 

groeperen ook op de database server plaatsvindt. Kijk naar de volgende SQL Query: 

 

 

 

Er zijn evenwel nogal wat voorwaarden waaraan een rapport moet voldoen voordat het groeperen door 

de database server wordt uitgevoerd. 

 

▪ Ten minste één groep 

▪ Detail sectie moet verborgen zijn 

▪ In de group header of footers mogen alleen groepsvelden of summary velden staan 

▪ Groepeer of selecteer niet op basis van Crystal Reports formula fields maar op SQL Expression fields 

▪ Running Totals mogen niet op Detailgegevens gebaseerd zijn, alleen op Summary fields 

▪ Het rapport mag geen Average of Distinct Count functies bevatten 

▪ Er mag alleen Asc of Desc gesorteerd zijn 

▪ de optie bij groeperen: Specified order werkt niet 

22.4 Geïndexeerde velden 

Om de performance zo hoog mogelijk te houden is het van belang zoveel mogelijk te linken op 

geïndexeerde velden: 

 

 

 

Klikken op de knop Index legend geeft een overzicht van de figuren: 

22.5 Volgorde van de relaties 

Wanneer we meer dan twee bestanden hebben gelinked, kunnen we op de knop Order links klikken. 

 

 

 

We krijgen dan het volgende beeld: 

 

 

 

In dit scherm kunnen we aangeven volgens welke volgorde de bestanden gelinked moeten worden en of 

die volgorde ook afgedwongen moet worden. Wel of niet afdwingen maakt het volgende verschil: 

 

A. Niet afgedwongen: oorspronkelijke volgorde wordt aangehouden 

B. Wel afgedwongen: gekozen volgorde wordt aangehouden 

 

 

 

Het een en ander kan uitmaken voor de snelheid waarmee de data worden opgehaald. 

 

22.6 Kleinste tabel voorop in link 

Over het algemeen is het aan te raden de kleinste tabel voorop te zetten. Vanuit de kleinste tabel hoeven 

minder links te worden gelegd en dat is dus sneller af te handelen.  

 

In bovenstaande voorbeeld zou Klanten dan voorop moeten staan gevolgd door Orders en ten slotte 

Orderinformatie 

 

We kunnen dat voorelkaar krijgen door in het Links scherm van de Database Expert eerst de tabellen in 

de juiste volgorde te zetten en daarna de links aan te brengen: 

Dubbelklikken op een link hoort dan dit te laten zien: 

 

 

 

Klanten staat voor Orders. 

 

We kunnen ook met de rechter muisknop klikken op één van de links en kiezen voor Reverse links 

De volgorde van de tabellen in de link wordt dan omgedraaid. 

 

22.7 Vermijd gekoppelde subrapporten 

Indien mogelijk moeten gekoppelde subrapporten niet gebruikt worden. Dit soort subrapporten hebben 

namelijk een heel slechte performance: de data moeten eerst worden opgehaald en kunnen daarna pas 

onderling gekoppeld worden. 

 

22.8 Opgave 

▪ In het volgende voorbeeld zullen we het verschil illustreren.  

▫ Maak een rapport met de tabellen klanten, orders en orderinformatie uit de database 

Noordenwind.  

▫ Maak een formula field Omzet  

 

 {Orderinformatie.Prijs per eenheid}*{Orderinformatie.Hoeveelheid} 

 

▫ Maak een SQL Expression field Omzet (ook via de field explorer) :  

 

 `Orderinformatie`.`Prijs per eenheid`*`Orderinformatie`.`Hoeveelheid` 

 

▫ Maak nu eerst een selectie maken op het eerste veld {@omzet} > 1000, dan ziet de SQL query 

(via menu database  show SQL Query) er als volgt uit: 

▫ sla het rapport op als oefening0101.rpt 

23 SQL Expression fields 

23.1 SQL Expression Field definitie 

SQL Expression fields lijken op formula fields, maar ze zijn geschreven in Structured Query Language 

(SQL). Ze zijn van belang voor het optimaliseren van de performace van een rapport doordat de 

opdracht die ze uitvoeren door de database server gedaan wordt. 

 

Je kunt SQL Expression fields gebruiken om voorgedefinieerde functies op database velden uit te voeren. 

De lijst van beschikbare functies hangt af van de database in gebruik. Deze lijst is beschikbaar in de 

Function Tree van de SQL Expression Editor.  

 

Let op: als Crystal Reports de functienaam herkent, wordt deze blauw in SQL Expression Editor' dialog 

box. Er kunnen andere database functies beschikbaar zijn, afhankelijk van de gekoppelde database, 

maar die worden niet blauw. 

23.2 SQL Expression field versus a Crystal Reports formula field 

Vaak kunnen we met een SQL Expression field als met een formula field. Stel bijvoorbeeld dat we de 

eerste drie letters van de naam van een werknemer willen afdrukken. Het voordeel van een SQL 

Expression field is dat de vraag door de database server wordt uitgevoerd in plaats van door Crystal 

Reports. Dat betekent een snellere uitvoering van het rapport.  

 

SQL Expression field: 

 

{fn LEFT("Employee"."First Name",3)} 

 

Formula field:  

 

left({Employee.First Name},3) 

23.3 SQL Expression Fields and SELECT Statements      

In het algemeen ondersteunt een SQL Expression field niet het gebruik van een SELECT statement omdat 

Crystal Reports maar één SELECT statement per  rapport kan uitvoeren. Als we een SELECT statement 

gebruiken in een SQL Expression field, kan de volgende foutmelding verschijnen: 

 

 

 

De exacte bewoordingen hangen af van de gebruikte database. Voor meer robuust gebruik van SQL 

kunnen we trouwens een Command gebruiken.      

Hoewel SQL Expressions met een SELECT statements niet ondersteund worden, kunnen ze toch werken 

als ze één waarde retourneren. Maximum, Minimum en Count zijn bijvoorbeeld functies die een enkele 

waarde retoruneren. Afhankelijk van de gebruikte database, kan het nodig zijn haakjes te gebruiken voor 

het SELECT statement. 

 

Microsoft SQL Server en Microsft Access vereisen bijvoorbeeld haakjes bij het SELECT statement. Het 

lijkt beter dit consequent te doen. 

 

Voorbeeld van de foutmelding voor de SQL Server als je geen haakjes plaatst: 

 

 

 

Voorbeeld met haakjes voor SQL Server: 

 

 ( 

 SELECT DISTINCT 'Yes' 

 FROM orders 

 WHERE CustomerId = "Customers"."CustomerID" 

 ) 

 

Zelfde voorbeeld voor Microsoft Access: 

 

 ( 

 SELECT DISTINCT 'Yes' 

 FROM orders 

 WHERE CustomerId = `Customers`.`CustomerID` 

 ) 

 

In beide gevallen ziet het Show SQL Query venster er ongeveer zo uit: 

23.4 Niet voorgedefinieerde Database Functies 

Database functies die niet voorgedefinieerd zijn in Crystal Reports kunnen toch gebruikt worden in SQL 

Expression fields. Bijvoorbeeld: 

 

 LOWER("table"."field") 

 

LOWER is geen voorgedefinieerde functie in Crystal Reports. Deze wordt dus niet blauw bij gebruik in 

de SQL Expression Editor. Toch zal dit geaccepteerd worden in het rapport. 

 

Nog een voorbeeld voor SQL Server: 

 
 DATEPART("WW", "Orders"."OrderDate") 

 

Een voorbeeld voor Microsoft Access: 

 
 DATEPART("WW", `Orders`.`OrderDate`,2,2) 

 
In beide gevallen zal het Show SQL Query venster er ongeveer zo uit zien: 

24 Grafieken 

In het volgende voorbeeld gebruiken we de tabellen Orderinformatie, Orders, Categorieën en 

Producten uit de database noordenwind. We willen een grafiek waarbij de gesommeerde omzet wordt 

afgezet tegen de categorienaam (kolom) en het kwartaal (orderdatum als rij). 

 

▪ We hebben weer het Formula field omzet (hoeveelheid * prijs) nodig 

▪ de rest kunnen we allemaal aanpakken via het menu Insert en de optie Graph 

▪ We plaatsen de grafiek in de Report Header 

 

We krijgen dan dit scherm: 

 

 

 

▪ Voor On change of kiezen we categorienaam en orderdatum uit de tabel orders 

▪ Bij orderdatum geven we dan via Order… aan dat we dat per kwartaal willen 

▪ Bij Show value(s) kiezen we bruto.  

▪ Als we dan op OK drukken krijgen we de volgende grafiek: 

 

 

De titel kunnen we achteraf aanpassen door er met de rechter muisknop op te klikken: 

Dit had ook vooraf gekund via het tabblad Text van de Chart Expert. Daar hadden we via Type ook een 

ander grafiektype kunnen kiezen.  

 

 

 

We kunnen achteraf ook het grafiektype van één serie aanpassen.  

▪ We klikken dan dan met de rechter muisknop op één van de series: 

 

 

▪ We kiezen dan Serie Options: 

 

 

 

 

24.1 Opgaven 

▪ Kies de tabellen Categorieën, Order, Orderinformatie en Producten uit de database Noordenwind 

▫ Maak het Formula Field Bruto (hoeveelheid * prijs) 

▫ Maak een groep  op Categorienaam 

▫ Voeg een grafiek toe en plaats deze in de kop van de groep Categorienaam 

▫ Kies als type Line 

▫ Zet Productnaam en Kwartaal af tegen Bruto 

▫ Sla het rapport op als oefening0201.rpt 

 

▪ Maak nog een grafiek met de tabellen klanten, orders en orderinformatie uit de database 

Noordenwind 

▫ We maken weer een Formula field bruto (hoeveelheid * prijs) 

▫ In een 3D grafiek zetten we klanten (verticaal) en jaren (horizontaal, op basis van orderdatum) 

tegen elkaar af 

▫ Op de knooppunten laten we het totaal van de bruto omzet berekenen 

▫ sla het rapport op als oefening0202.rpt 

 

▪ En ten slotte een heel ingewikkelde opgave. We willen een grafiek waarbij de waarden onder en 

boven de norm een andere kleur hebben; de normlijn zelf nemen we ook op 

▫ We maken een grafiek op basis van de orders en de orderinformatie.  

▫ We willen nu een grafiek die de omzet over de maanden weergeeft waarbij de waarden die 

onder het gemiddelde liggen een andere kleur moeten krijgen dan de waarden die erboven 

liggen 

▫ probeer dit aan de hand van het voorbeeld voor elkaar te krijgen 

 

 

 

▫ En als dat niet lukt geven we de oplossing 

▫ We maken een Formula field omzet dat de prijs per orderregel berekent (hoeveelheid * prijs) 

▫ We maken een formula field maand dat de maand uit de orderdatum haald met de functie 

month 

▫ We groeperen op maand 

▫ We berekenen het totaal van omzet 

▫ En plaatsen dit in de group header van de groep maand 

▫ We berekenen in het Formula field gemomzetperjaar ook het gemiddelde van de omzet per 

maand door het totaal van de omzet te delen door het aantal maanden 

 

 Sum ({@omzet})/DistinctCount ({@maand}) 

 

▫ We maken nog twee formula fields omzetboven en omzetonder 

▫ In omzetboven geven we de omzet als deze boven het gemiddelde ligt 

 

 

 

 if Sum ({@omzet},  {@maand}) > {@gemomzetperjaar} then 

    Sum ({@omzet},  {@maand}) 

 else 

    0 

 

▫ In omzetonder doen we het tegenovergestelde 

 

 if Sum ({@omzet}, {@maand}) <= {@gemomzetperjaar} then 

    Sum ({@omzet},{@maand}) 

 else 

    0 

 

▫ Nu maken we een grafiek die we in de report header plaatsen, met de volgende keuzes 

▫ We klikken op OK 

▫ De grafiek ziet er in eerste instantie heel anders uit dan de bedoeling is 

 

 

 

▫ Nu kiezen we via de Chart Expert als type Line 

▫ De lijn voor het gemiddelde laten we zoals deze is 

▫ De andere twee veranderen we in type Riser 

▫ We selecteren de betreffende lijn 

▫ En klikken er rechts op 

▫ En kiezen Series Options 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 154 

 

 

▫ Rechts bovenin kiezen we Riser 

▫ We doen dat voor beide lijnen 

▫ We krijgen dan 

 

 

o Sla het rapport  op als oefening0203.rpt 

25 Crystal Reports Templates 

Crystal Reports templates zijn gewoon bestaande RPT bestanden die gebruikt worden om andere 

rapporten op te maken. 

 

We kunnen deze templates toepassen via het menu Report en de optie Template Expert: 

 

  

 

De templates bevinden zich in een map van Crystal Reports 

 

 

 

Als je zelf een template wil maken, kun je dat dus doen door simpelweg een rapport op te slaan in de 

betreffende template map.  

Er is daarbij één extra hulpmiddel: het Template Field Object. Dit is te vinden onder het menu Insert. 

 

Als je een template field hebt ingevoegd, zijn er veel meer mogelijkheden dit op te maken dan normaal. 

Als je op het veld klikt met de rechter muisknop, krijg je het volgende scherm: 

 

 

 

Als je bijvoorbeeld niet weet of het eerste veld in de detail sectie een string, numeriek of datum gegeven 

zal zijn, kun je respectievelijk de Paragraph, Number of Date and Time tab gebruiken. 

 

25.1 Opgaven 

▪ Maak een simpel rapport op basis van de tabel Orderinformatie uit de database Noordenwind 

▫ groepeer op order-id 

▫ plaats een aantal gewone en template velden in de detailsectie 

▫ Pas op dit rapport de template Corporate_Blue_EN.rpt toe via Reportt en de optie Template 

Expert 

▫ Open ook de oorspronkelijke template en onderzoek wat er precies is toegepast 

▫ sla het rapport op als oefening0301.rpt 

 

▪ Probeer zelf eens een simpel template te maken - met een aantal template velden  

▫ Pas dit vervolgens toe op een niet opgemaakt rapport 

▫ sla het rapport op als oefening0302.rpt 

26 Multipass rapport 

26.1 Inleiding 

Met pass of fase wordt een proces bedoeld dat door Crystal Reports wordt uitgevoerd zodra er gegevens 

worden gelezen of bewerkt. Afhankelijk van de mate van complexiteit van het rapport worden de 

gegevens door Crystal Reports in 1, 2 of 3 fasen verwerkt. Dankzij deze werkwijze kunt u complexe 

rapporten en formules gebruiken. 

 

De uitkomst van formules die variabelen bevatten wordt beïnvloed door de feitelijke plaats binnen het 

rapport. Wil je bijvoorbeeld waarden checken en variabelen vullen terwijl de records één voor één 

doorlopen worden, dan moet de formule in de detailsectie. Om de variabelen te resetten, moet er een 

formule in de kop van de groep. 

 

Het plaatsen van de formules in de juiste secties garandeert evenwel nog niet dat ze op het juiste 

moment (in de juiste pass) geëvalueerd worden. Dat hangt namelijk ook nog van het soort formule af. 

 

In het onderstaande voorbeeld hebben we de tabellen Orders en Customers uit de database Xtreme2000 

gebruikt.  

 

▪ In dit voorbeeld berekenen we zelf een Running Total met een formule: 

 

CurrencyVar MonthlyTotal := MonthlyTotal + {Orders.Order Amount} 

 

 

 

Dat levert het juiste resultaat op! 

 

▪ Vervolgens gaan we groeperen op de orderdatum, per maand. 

▪ En dan krijgen we een volslagen onlogisch resultaat. Ook het resetten van het totaal heeft geen 

effect.  

▪ We plaatsen de formule:  

 

 CurrencyVar MonthlyTotal :=0; 

 

▪ In de kop van de groep en er verandert er niets. 

 

Wat is hier aan de hand? Er zijn drie zogenaamde passes in Crystal Reports. In elk van deze passes 

vinden bewerkingen plaats: 

 

Before Reading Records  Vindt plaats voordat er records uit de database gelezen 

zijn (flat formulas) 

While Reading Records Vindt plaats als de records uit de database gelezen 

worden; formules die directe verwijzingen ( maar geen 

accumulaties) naar velden bevatten, worden in deze 

fase berekend (first pass formulas) 

While Printing Records Vindt plaats als de records gelezen zijn en klaar voor 

afdrukken; formules die accumulaties bevatten, worden 

nu berekend (second pass formulas) 

 

Ons Running Total is een zogenaamde first pass formula. Daarna worden de bedragen anders 

gerangschikt door het groeperen. Het Running Total lijkt nu volslagen onlogisch.  

 

De reset formule is een flat formula. Die wordt dus bij de start van het rapport uitgevoerd en heeft geen 

effect op het Running Total.  

 

Hoe lossen we dit op? Ons Running Total laten we in een latere pass uitvoeren door een regel toe te 

voegen: 

 

 WhilePrintingRecords; 

 CurrencyVar MonthlyTotal := MonthlyTotal + {Orders.Order Amount} 

 

Het resetten werkt dan nog steeds niet. Logisch. Ook de reset formula moeten we in een latere fase laten 

uitvoeren 

 

 whileprintingrecords; 

 CurrencyVar MonthlyTotal :=0 

 

En dan werkt alles correct! 

 

Let op:  we kunnen formulas alleen in een latere pass of fase laten uitvoeren en logischerwijs niet in 

een eerdere. Zouden we dat toch proberen, dan krijgen we een foutmelding. 

26.2 Opgaven 

▪ Maak een vergelijkbaar voorbeeld op basis van de noordenwind database gegroepeerd per kwartaal 

op orderdatum en met de netto omzet 

▫ sla het rapport op als oefening0401.rpt 

 

▪ Maak een rapport op basis van de tabel klanten.  

▫ Gebruik uit de tabel klanten de kolom bedrijf.  

▫ Groepeer op de eerste letter van het land. 

▫ Maak drie tellers: teller1, teller2 en teller 3 met als inhoud: 

 

 numbervar teller1:=teller1+1 

 

▫ Plaats in de Group header formules resetteller1 etc. met als inhoud: 

 

 numbervar teller1:=0 

 

▫ Experimenteer met de commando’s Whilereadingrecords en Whileprintingrecords bij de 

laatste twee tellers en de laatste twee reset formules. 

 

 

 

▫ sla het rapport op als oefening0402.rpt 

27 Reverse engineering van bestaande rapporten 

Uit Wikipedia: Reverse engineering is het onderzoeken van een product (meestal software of 

een communicatieprotocol) om daaruit af te leiden wat de eisen zijn waaraan het product probeert te 

voldoen, of om de precieze interne werking ervan te achterhalen. 
 

Hoe kunnen we van een bestaande rapport achterhalen hoe het is opgebouwd. Zelf volg ik daarbij de 

volgende stappen. 

27.1 Report definition 

Bijzondere aandacht verdient hierbij de export optie Report Definition.  

 

Met deze optie is het mogelijk een gedetailleerde beschrijving van het rapport naar een tekstbestand te 

laten wegschrijven. Zie voorbeeld: 

 

http://nl.wikipedia.org/wiki/Software
http://nl.wikipedia.org/wiki/Protocol#communicatieprotocol

 Crystal Report Professional v10.0 (32-bit) - Report Definition 

 

 1.0 File Information 

 

 Report File:  

 Version: 10.2 

 

 2.0 Record Sort Fields 

 

 3.0 Group Sort Fields 

 

 4.0 Formulas 

 

 4.1 Record Selection Formula 

 {Leveranciers.Land} = "Canada" 

 

 4.2 Group Selection Formula 

  

 4.3 Other Formulas 

 

 5.0 Sectional Information 

 

 5.1 Page Header Section 

 Visible, Keep Together 

 

 Bedrijf 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 Plaats 

  String, Visible, Left Alignment, Top Alignment,  

  Keep Together 

 

 5.2 Page Footer Section 

 Visible, New Page After, Keep Together, Print At Bottom of Page 

 

 5.3 Report Header Section 

 Visible, New Page Before 

 

 5.4 Report Footer Section 

 Visible, New Page After 

 

 5.5 Group Header Section 

 Visible 

 

 GroupName ({Leveranciers.Land}) 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 

 5.6 Group Footer Section 

 Visible 

 

 5.7 Details Section 

 Visible 

 

 Subsection.1  

  Visible, Keep Together 

 

 {Leveranciers.Bedrijf} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

 {Leveranciers.Plaats} 

  String, Visible, Default Alignment, Top Alignment,  

  Keep Together, Using System Default Formatting, Word Wrap 

27.2 Gebruikte koppeling en tabellen 

Via de database expert zoeken we uit welke koppeling en tabellen er gebruikt zijn. 

 

 

 

Wat betreft de tabellen moeten we ook nog naar de mogelijkheid kijken of er zogenaamde Commands 

c.q. SQL view tussen zitten: 

 

 

 

Bijvoorbeeld: 

 SELECT naam, AantalCompetitieRonden FROM vtcompetities WHERE id = 

{?CompetitieID} 

27.3 Links 

Via dezelfde database expert bepalen we wat de links zijn: 

 

 

 

We kunnen de tabellen tijdelijk kleiner maken door dubbel te klikken op de titelbalk: 

 

 

 

Voor elke link moeten we vaststellen wat het karakter van die link is, via Link Options: 

27.4 Show SQL Query 

Via het menu Database en de optie Show SQL Query kunnen we de SQL instructie achterhalen die 

Crystal Reports naar de database stuurt: 

 

 

27.5 Filter 

De volgende stap is het vaststellen van het Filter: 

27.6 Groene vinkjes bij gebruikte velden en formula fields 

In de Field Explorer zien we groene vinkjes bij gebruikte velden en Formula fields: 

 

 

 

Bij Parameter Fields ontbreekt dit. 

27.7 Rechter muisknop bij Fields, Formula fields en Parameter Fields 

Vanaf versie 2008 zien we dat we via de rechter muisknop kunnen achterhalen in welke Formula Fields 

het betreffende veld, Formula Field of Parameter Field voorkomt: 

We krijgen dan bijvoorbeeld: 

 

 

27.8 Opgaven 

▪ Pas het een en ander toe op het rapport treinmeerderejaren.rpt 

▫ Maak een Word document treinmeerderejaren.doc(x) 

▫ Maak een export van de report definition van het rapport  

▫ Kopieer de tekst naar hoofdstuk 1 

▫ Maak een overzicht van de gebruikte tabellen en de onderlinge relaties 

▫ Plak dit overzicht in hoofdstuk 2 van het Word document 

▫ Bepaal van elke link het karakter, left join, right join etc. 

 

 

 

▫ Plaats deze plaatjes eveneens in hoofdstuk 2 

▫ Roep de SQL query op via Database 

▫ Plak dit plaatje in hoofdstuk 3 van het Word document 

▫ Roep het filter op  

▫ Plak het plaatje in hoofdtuk 4 van het Word document 

▫ Bepaal via de Field Explorer welke velden, Formula Fields etc. in het rapport gebruikt worden 

(te zien aan de groene vinkjes) 

 

 

 

▫ Maak afdrukken 

▫ Plak deze in hoofdstuk 5 van het Word document 

▫ Bepaal via de rechter muisknop waar velden gebruikt worden in Formulas 

▫ Maak van een aantal afdrukken 

▫ Plak deze in hoofdstuk 6 van het Word document 

▫ Sla het Word document op als oefening0501.doc(x) 

28 Geavanceerd formule gebruik 

In de basiscursus is al uitvoerig met simpele formules gewerkt. In dit hoofdstuk willen we een aantal 

voorbeelden geven van geavanceerder gebruik. We zullen trouwens bij deze voorbeelden ontdekken dat 

de moeilijke constructies vaak ook te vervangen zijn door simpeler varianten met behulp van een 

bestaande functie. 

28.1 Bijhouden hoogste orderregelsbedrag, datum en klant: if then 

Voor het navolgende voorbeeld gebruiken we de tabellen Orderinformatie, Order en Klanten uit de 

database noordenwind. We willen in het rapport zien welke klant op welke datum de orderregel met het 

hoogste bedrag heeft. 

 

▪ We maken een Formula Field met de volgende inhoud: 

 

 currencyvar BonusAmount; // zorgt er voor dat BonusAmount afgebeeld 

wordt als bedrag!!! 

 stringvar HighestCustname; 

 Datetimevar DateBonusReached; 

 

 if {@bruto omzet}> BonusAmount then 

    ( 

    HighestCustname := {Klanten.Bedrijf}; 

    DateBonusReached := {Orders.Orderdatum}; 

    BonusAmount := {@bruto omzet} 

    ); 

 //let op de haakjes!!!!! 

 "Het hoogste bedrag is  " & totext(BonusAmount) & " door " & 

HighestCustname & "op " & 

 totext(DateBonusReached,"d/M/yyyy") 

 

▪ Dit veld plaatsen we in de Report footer 

▪ De overige secties verbergen we.  

 

Het eindresultaat moet er dan als volgt uitzien: 

 

Het antwoord klopt trouwens niet. Er zijn twee klanten met een grootste orderregel van 15.810. We 

kunnen dit antwoord tevoorschijn krijgen met behulp van de functie Nthlargest (categorie Summary). De 

grammatica van deze functie is als volgt: NthLargest (10,{@bruto} ). Dit geeft aan welke bruto bedrag de 

tiende grootste is. 

28.2 Streepjes uit telefoonnummer halen: While Do loop 

In het volgende voorbeeld gebruiken we een While Do loop om net zolang van links naar rechts door 

een telefoonnummer te lopen totdat alle streepjes er uit zijn. We hebben de tabel Customers uit de 

database Xtreme2000 gebruikt.  

 

In het veld Phone staan meerdere streepjes. Het Formula Field om die streepjes eruit te halen ziet er zo 

uit: 

 StringVar NewPhone := {Customer.Phone}; 

 

 While Instr(NewPhone,"-")>0 Do 

 ( 

    NewPhone := Left(NewPhone, Instr(NewPhone,"-")-1) & 

    Right(NewPhone, Length(NewPhone) - Instr(NewPhone,"-")); 

 ); 

 NewPhone 

 

Het rapport ziet er zo uit: 

 

 

Mooi voorbeeld van een While Do loop, maar hetzelfde hadden we ook kunnen doen met de replace 

functie.  

28.3 Formulas om werkdagen tussen twee data te berekenen 

Voor het onderstaande voorbeeld hebben we de tabel Orders uit Noordenwind gebruikt. We willen het 

verschil in werkdagen berekenen tussen de Orderdatum en de Leverdatum. De oplossing bestaat uit 

twee delen. In het eerste deel nemen we de eventuele feestdagen. 

 

 //Vakantiedagen, moet in de report header van het rapport 

 DateVar Array Holidays := [ 

 Date (1996,07,10),  

 Date (1997,12,25), 

 Date (1997,12,31)  

 ]; 

 0 

 
In het tweede deel berekenen we hoeveel werkdagen er tussen twee datums zitten waarbij we er eerst 

de zater- en zondagen vanaf trekken. Vervolgens kijken we of er van de de tussenliggende dagen ook 

dagen bij de feestdagen horen. Die trekken we vervolgens ook nog af van het totaal. 

 



Crystal Reports cursus I 

© H. de Walle (www.walmar.nl)  pagina 172 

 //Hoofd formula, in de detailsectie 

 Local DateVar Start := datevalue({Orders.Orderdatum});  

 Local DateVar End := datevalue({Orders.Leverdatum});  

 Local Numbervar Hol; 

 DateVar Array Holidays; 

 

 numbervar diff; 

 diff:=datediff("d",Start,End)- 

 DateDiff ("ww", Start, End, 7) – 

 DateDiff ("ww", Start, End, 1); 

 

 Local NumberVar i; 

 For i := 1 to Count (Holidays) 

 do (if DayOfWeek ( Holidays[i] ) in 2 to 6 and  

 Holidays[i] in start to end then Hol:=Hol+1 ); 

 diff - Hol 

28.4 Opgaven 

▪ Maak een nieuw rapport met de tabellen orders en orderinformatie uit de database noordenwind 

▫ Maak een formula field bruto dat de velden [prijs per eenheid] en [hoeveelheid] met elkaar 

vermenigvuldigt. 

▫ Maak een formula field top10 met als inhoud: NthLargest (10,{@bruto} ) 

▫ Plaats beide velden in de detailsectie 

▫ Verberg de detailsectie via de section expert met als voorwaarde  

 

 {@bruto}<{@top10} 

 

▫ sla het rapport op als oefening0601.rpt 

 

▪ Maak een nieuwe rapport met de tabel leveranciers uit de database Noordenwind 

▫ Probeer aan de hand van het bovenstaande voorbeeld een formula field te maken dat alle 

spaties uit het veld [bedrijf] te halen.  

▫ sla het rapport op als oefening0602.rpt 

▫ probeer het probleem ook op te lossen met een formula field dat de Replace functie gebruikt 

▫ gebruik de helpfunctie om uit te vinden hoe je dat met deze functie moet aanpakken 

▫ sla het rapport op als oefening0603.rpt 

29 Gebruik van Arrays 

29.1 Inleiding 

Een array (Engels voor rij of reeks) is bij het programmeren van computers een variabele c.q. 

datastructuur die bestaat uit een lijst van elementen. Ieder element in een array heeft een unieke index 

waarmee dat element aangeduid kan worden. 

 

Een mooi voorbeeld hiervan in Crystal Reports is de parameter met meerdere waarden. Stel we hebben 

in een rapport met de table Klanten een parameter test die een keuze uit landen mogelijk maakt. 

 

▪ We kiezen België, Canada, Duitsland 

▪ We maken een Formula field test die verwijst naar deze parameter met een indexnummer 

 

 {?test}[2] 

 

▪ Als we deze formula in het rapport plaatsen, krijgen we de waarde Canada 

29.2 Voorbeeld: dynamische arrays 

Vraag: 

Ik wil een lijst van details of groepen combineren  en die printen in de footer. Hoe krijg ik dat voor 

elkaar? 

 

Antwoord: 

Dat kan door in Crystal Reports formulas dynamische arrays te gebruikern. 

Er zijn twee typen arrays in Crystal Reports – Static and Dynamic. Eéndimensionale arrays kunnen een 

maximum van 1000 elementen bevatten. 

 

Om een array te gebruiken, moet je eerst array declareren en er dan waarden aan toeschrijven. Ten 

slotte verwijs je naar de elementen. 

 

Arrays hebben dezelfde scope als datatypes en andere  variabelen. Ze bestaan uit: 

  

▪ Array naam 

▪ Assignment Operator 

▪ List met waarden 

 

 Scope Datatype  Array Name:= [value1, value2, value2,  etc…]; 

 

We verwijzen naar de waarde van een array met een cijfer 

 

 Name[x] 

 

Waarden veranderen in een array   

 

 Name[x] := nieuwe waarde; 

 

In het volgende voorbeeld heeft een klant een lijst van verschillende producten. De volledige lijst moet 

op één regel geprint worden. Dat kunnen we doen door Dynamic Array. Een array wordt gevuld met 

een lijst van van producten en geprint in de groep footer. 

Voor dit voorbeeld hebben we drie formula fields nodig: 

 

▪ Initialize Array 

▪ Load Array 

▪ Print Array 

 

1. Initialize Array plaatsen we in de Customer group header. 

 

 Shared stringVar array productType := [""]; 

 Shared numberVar x := 1; 

2. Load Array plaatsen we in de Product Type group header. 

 

 Shared numberVar x; 

 Shared stringVar array productType; 

 productType [x] := {Product_Type.Product Type Name};  //Load array with 

product type 

 Redim preserve productType[UBound(productType)+1]; //more space added 

to array. 

 x := x+1; 

 

3. Print Array plaatsen we in Customer group footer. 

 

 Shared stringVar array productType; 

 Join (productType, ", ") // Use Join function to print array on one 

line 

 

Nu kunnen we de lijst van producten op één regel afdrukken. 

29.3 Voorbeeld: Array als filter 

We kunnen een Array bijvoorbeeld gebruiken om de unieke waarden van een bepaald veld eruit te 

filteren, terwijl we wel alle velden willen laten zien. 

 

In het navolgende voorbeeld maken we gebruik van de tabellen Customers en Orders uit de database 

Xtreme2000.  

 

We maken we gebruik van een Array van 1000 elementen, het maximum aantal dat een Array in Crystal 

Reports kan bevatten.  Willen we meer, dan zullen we een tweede Array moeten maken enzovoorts. 

 

 

 

Elke keer als een record gelezen wordt, wordt het ordernummer toegevoegd aan de Array, mits het er 

nog niet inzit. De formule luidt: 

 //{@DynamicArrayBuilder} 

 whileprintingrecords; 

 numbervar array MyArray; 

 numbervar Counter; 

 // de regel hieronder zorgt er voor dat alleen nieuwe elementen aan de 

array worden toegevoegd 

 if not({Orders.Order ID} in MyArray) then 

 // de code hieronder wordt geactiveerd als de veldwaarde nog niet in de 

array staat 

 // de teller wordt met één opgehoogd 

 ( 

 Counter := Counter + 1;  

 // de regel hieronder voorkomt dat de array meer dan 1000 elementen 

gaat bevatten 

 if Counter <= 1000 then 

 // met Redim wordt de omvang van de array opgehoogd tot de nieuwe 

waarde van de Counter 

 // Preserve dat de bestaande elementen van de Array op hun plaats 

blijven 

    ( 

    Redim Preserve MyArray[Counter];  

    // de nieuwe waarde wordt toegevoegd aan het nieuwe element van de 

array 

    MyArray[Counter] := {Orders.Order ID} 

    ) 

 ); 

 

Deze formule wordt in de detailsectie B geplaatst. In de detailsectie A staat een formule om te tonen 

welk ordernummer er is toegevoegd. De toevoeging WhilePrintingRecords is nodig om beide formulas in 

dezelfde pass te laten plaatsvinden.  

 

 // deze formula is niet vereist om dynamische arrays te produceren; 

wordt  alleen gebruikt 

 // om Array info te tonen 

 whileprintingrecords; 

 numbervar array MyArray; 

 numbervar Counter; 

 

 if {Orders.Order ID} in MyArray then  

  "already in array" 

 else  

  "added to array, element " + totext(Counter + 1,0) 

 

Deze formule moet in een vroegere sectie staan omdat anders de conclusie steeds zou zijn dat het 

ordernummer al in de Array zit. Om de sectie toch om gelijke hoogte te krijgen met de volgende is 

Underlay Following Sections via de Section Expert aangevinkt. 

Er wordt een teller bijgehouden om aan te geven hoeveel elementen er aan de Array zijn toegevoegd.  

 

 // deze formula is niet vereist; hij toont alleen de ophoging van de 

Array 

 numbervar Increment; 

 if onfirstrecord or {Orders.Order ID} <> previous({Orders.Order ID}) 

then  

  Increment:= Increment + 1; 

 Increment 

 

Deze teller blijft trouwens doorlopen nadat de duizend elementen bereikt zijn. 

 

Met onderstaand formula field worden alle waarden uit de numeriek Array uitgelezen en in een 

variabele gezet. Deze variabele plaatsen we in de Report Footer 

 

 whileprintingrecords; 

 numbervar i:=1; 

 stringvar totaalarray; 

 numbervar array MyArray; 

 for i := 1 to count(MyArray) do 

    ( 

    if i < count(MyArray) then 

        totaalarray := totaalarray & totext(MyArray[i],0,"") & ", " 

    else 

        totaalarray := totaalarray & totext(MyArray[i],0,"") 

    ); 

 totaalarray; 

29.4 Opgaven 

▪ Maak het bovenstaande voorbeeld na 

▫ sla het rapport op als oefening0701.rpt 

▪ Maak een vergelijkbaar voorbeeld op basis van de noordenwind database met de tabellen Orders 

en Klanten 

▫ Groepeer op Bedrijfsnaam en zorg dat de ordernummers verzameld worden via een Array en 

samen met de Bedrijfsnaam allemaal in de Group Footer getoond worden 

▫ We hebben allereerst een Formula Field arrayorders 

 

 whileprintingrecords; 

 stringVar array ordernummers; 

 numbervar i; 

 i:=i+1; 

 redim preserve ordernummers[i]; 

 ordernummers[i]:= totext({Orders.Order-id},0,"") 

 

▫ Vervolgens maken we leesarray 

 

 whileprintingrecords; 

 stringVar array ordernummers; 

 join(ordernummers,", ") 

 

▫ Ten slotte maken we resetarray 

 

 whileprintingrecords; 

 numbervar i:=0; 

 

▫ Plaats het een en ander zo: 

 

 

 

▫ Zo komt het er dan uit te zien 

 

 

 

▫ Sla het rapport op als oefening0702.rpt 

30 Subrapporten 

30.1 Hoofdrapport-subrapport 

Stel we willen de volgende vraag beantwoorden. We willen van de producten die uit het assortiment 

genomen zijn, weten wie de klanten waren, wie de leverancier en de omzet over een bepaalde periode. 

We hebben dan subrapporten nodig. 

 

▪ Voor het hoofdrapport gebruiken we de tabel Producten uit de database Noordenwind 

▪ We hoeven alleen de productnamen maar te zien 

▪ We filteren de producten eruit die uit het assortiment zijn:  

 

 

 

▪ Het komt er zo uit te zien: 

 

 

 

▪ Het eerste subrapport bevat de besdrijfsnamen (een koppeling met producten moet mogelijk zijn: 

de tabellen orders, orderinformatie en producten moeten ook opgenomen worden) 

▪ We groeperen op Bedrijf om te zorgen dat we alleen unieke bedrijven te zien krijgen 

▪ De details verbergen we 

▪ Vervolgens maken we het subrapport Leveranciers.  

▪ Hier hoeven we alleen de tabel Leveranciers maar te gebruiken 

▪ Wel moeten we ook hier groeperen op leverancier 

▪ We verbergen weer de details 

▪ Het derde subrapport bevat de bruto omzetcijfers per jaar per product. 

 

De drie laatste rapporten koppelen we aan het eerste: 

 

 

 

Via het link-scherm geven we aan hoe ze gelinkt moeten worden: 

Uiteindelijk komt het rapport er dan zo uit te zien: 

 

 

30.2 Opgaven 

▪ Maak een hoofdrapport op basis van de tabel producten uit de database Noordenwind 

▫ Plaats de productnaam in de detailsectie 

▫ Maak twee aparte rapporten: één met alle leveranciers (opslaan als: leveranciers)  

▫ en één met de total bruto omzet per jaar (opslaan als: omzet per jaar) 

▫ Plaats de twee gemaakte rapporten als subrapporten in het hoofdrapport  

▫ en link het eerste  op basis van lever-id en het tweede op productnummer 

▫ sla het rapport op als oefening0801.rpt 

 

▪ We willen nu hetzelfde overzicht per product alleen dan met parameter jaar om te kunnen kiezen 

voor welk jaar we een overzicht willen. Het verschil met de vorige opgave is dat het hoofdrapport 

een parameter jaar moet bevatten en het subrapport met de omzetcijfers een Formula Field jaar met 

als formule: year({orders.orderdatum}).  

De link moet bovendien dubbel zijn: niet alleen op productnummer maar ook op de parameter 

jaar, gelijk aan het Formula Field jaar uit het subrapport:  

▫ Maak in het hoofdrapport een parameter jaar en een formula field met als inhoud de uitkomst 

van de parameter 

▫ Maak in het subrapport Omzet per jaar een formula field jaar 

▫ Link hoofdrapport en subrapport Omzet per jaar nu dubbel op basis van productnummer en 

jaar 

▫ sla het rapport op als oefening0802.rpt 

30.3 Koppelen op basis van Shared variables 

Als het om één of andere reden niet mogelijk is te linken via het doorgeven van parameters, is er nog 

een andere methode: het gebruik van Shared variables. 

 

In het onderstaande voorbeeld gebruiken we de tabellen Producten en Klanten uit Noordenwind. 

 

In het rapport producten maken we een een formule met een Shared Variable: 

 

 shared numbervar productnummer:={Producten.Productnummer}; 

 

Dit veld plaatsen we in het de detail sectie van het rapport naast de productnaam. 

Vervolgens maken we een subrapport Leveranciers met daarin ook een formule met Shared Variable met 

dezelfde naam als in het hoofdrapport: 

 

 shared numbervar productnummer; 

 

Vervolgens onderdrukken we in het subrapport alle regels waarvoor geldt dat het productnummer niet 

gelijk is aan de Shared Variable: 

 

 {Leveranciers.Productnummer}<>{@productnummer} 

 

Feitelijk koppelen we dus niet maar tonen we domweg alleen de records die voldoen aan het 

doorgegeven criterium. 

30.4 Koppelen op basis van deelsleutels en doorgeven van waarden 

Het volgende verhaal is tamelijk ingewikkeld en vooral bedoeld als illustratie! 

 

Koppelen op basis van deelsleutels is niet zo’n probleem. We maken in het subrapport een Formula 

Field waaraan we de sleutel aan het hoofdbestand kunnen koppelen. Veel lastiger is het waarden door te 

geven van het subrapport aan het hoofdrapport. 

 

Het volgende voorbeeld is gebaseerd op de tabellen deelsleutel en completesleutel uit de database 

Voorbeeld. We willen deze bestanden aan elkaar koppelen met behulp van subrapporten en dan een 

waarde doorgeven van het subrapport naar het hoofdrapport.  

 

In het subrapport, gebaseerd op de tabel completesleutel maken we een formula field  

 

 deelsleutel: left({sleutel},3) 

 

Het hoofdrapport, gebaseerd op de tabel deelsleutel, linken we aan dit subrapport: 

Let op:  als je in het subrapport de optie Save data with report aan hebt staan, gaat het linken niet 

goed! 

 

In het subrapport maken we vervolgens een Formula field doorgeef: 

 

 shared stringvar test; //SHARED is nodig voor het doorgeven van hoofd- 

naar subrapport v.v. 

 test := {completesleutel.omschrijving}; 

 test 

 

Het Formula Field doorgeef moeten we opnemen in de detailsectie! Anders kan de waarde niet worden 

doorgegeven. Het subrapport komt er nu zo uit te zien: 

 

 

 

In het hoofdrapport maken we ook een Formula Field doorgeef: shared stringvar test, en een Formula 

Field reset:  

 

 shared stringvar test:="".  

▪ We maken in het hoofdrapport drie detailsecties: A, B en C.  

▪ In A plaatsen we Reset  

▪ In B. het subrapport  

▪ En in C Doorgeef. 

 

Het rapport komt er dan zo uit te zien: 

 

▪ Als we het een ander nog verder aan passen, door bijvoorbeeld bij Detailsectie A en B Underlay 

Following Sections aan te vinken: 

 

 

 

Dan krijgen we uiteindelijk het volgende beeld: 

30.5 Opgaven 

▪ Probeer zelf het voorbeeld in paragraaf 8.3 na te maken 

▫ Sla het rapport op als 0804.rpt 

31 Repository 

De Repository Custom Functions maakt het mogelijk diverse objecten zoals Custom Functions, 

Commands, Images en Text Objects, centraal op te slaan in een database. Daarmee worden ze 

beschikbaar gemaakt voor andere rapporten. 

 

In versie 9 (toen de Repository werd geïntroduceerd) hebben we een stand alone database; in de latere 

versies hebben we Business Objects Enterprise + license keys nodig. 

 

 

 

Zonder deze centrale opslag beginnen we dus niet zo veel. Wat we nog wel kunnen is Custom 

Functions maken en die centraal in het rapport opslaan. 

31.1 Custom functions 

We willen een rapport op basis van de tabellen orders en orderinformatie uit de database noordenwind. 

We willen groeperen op orderdatum en per kwartaal. De getoonde kwartalen moeten dan automatisch 

omgezet worden van 7 – 1996 naar 1996 3
de

 kwartaal. We willen dit omzetten doen met een 

zogenaamde Custom Function. 

 

▪ Custom functions maken we door in het menu Report te kiezen voor Formula Workshop.  

▪ We krijgen dan het volgende scherm: 

 

 

 

▪ Door met de rechter muisknop op Report Custom Functions te klikken en New … te kiezen, 

kunnen we een nieuwe functie maken.  

▪ We maken er ééntje die we kwartaal noemen. 

▪ We maken dan de volgende functie: 

 

 Function (datevar datum) 

    select month(datum) 

    Case 1 to 3:  

        totext(year(datum),0,"") + " " + "1ste kwartaal" 

    Case 4 to 6:  

        totext(year(datum),0,"") + " " + "2de kwartaal" 

    Case 7 to 9:  

        totext(year(datum),0,"") + " " + "3de kwartaal" 

    Case 9 to 12:  

        totext(year(datum),0,"") + " " + "4de kwartaal" 

 

 

 

▪ Als we deze functie nu willen gebruiken binnen hetzelfde rapport, dan vinden we hem terug onder 

Custom functions: 

 

 

 

▪ We kunnen deze functie nu toepassen op de Orderdatum:  

 kwartaal (datevalue({Orders.Orderdatum})).  

 

De functie datevalue is hier nodig voor de casting van de datum. Anders krijgt de functie kwartaal niet 

de juiste input, namelijk een echte datum. 

 

De zojuist gemaakte functie is nu overal binnen dit zelfde rapport aan te roepen, maar nog niet in 

andere rapporten. Willen we dit, dan moeten we de gemaakte functie toevoegen aan de Repository.  

 

▪ We doen dit door met de rechter muisknop op de functie te klikken: 

 

 

 

▪ We kiezen dan voor Add to Repository. 

 

Opgaven 

▪ Werk bovenstaande voorbeelden zelf uit 

▫ Maak een functie die uit een datum automatisch jaar en maandnaam in het Nederlands haalt. 

 

 Function (datevar datum) 

 totext(year(datum),"0000") & " " & 

 select month(datum) 

    case 1: "januari" 

    case 2: "februari" 

    case 3: "maart" 

    case 4: "april" 

    case 5: "mei" 

    case 6: "juni" 

    case 7: "juli" 

    case 8: "augustus" 

    case 9: "september" 

    case 10: "oktober" 

    case 11: "november" 

    case 12: "december" 

 

▫ Pas deze functie toe op orderdatum en leverdatum in de tabel Orders in Noordenwind 

 

 jaarmaandnaam (datevalue({Orders.Orderdatum})) 

▫ Sla het rapport op als oefening0901.rpt 

 


